Arithmetic Algorithms for Hereditarily Binary Natural Numbers II

Paul Tarau

Department of Computer Science and Engineering
University of North Texas
tarau@cs.unt.edu

Abstract. Our tree based *hereditarily binary numbers*, apply recursively a run-length compression mechanism that enables computations limited only by the structural complexity of their operands rather than by their bitsizes.
This paper describes several new arithmetic algorithms on hereditarily binary numbers and confirms empirically that, while within constant factors from their traditional counterparts for their worst case behavior, arithmetic operations on hereditarily binary numbers make tractable important computations impossible with traditional number representations.
Keywords: *hereditary numbering systems, compressed number representations, arithmetic computations with giant numbers, tree-based numbering systems, compact representation of large prime numbers*

1 Introduction

This paper is a sequel to [1][1] where we have introduced a tree based number representation, called *hereditarily binary numbers*, that uses *run-length encoding of bijective base-2 numbers*, recursively.

[1] presents specialized algorithms for basic arithmetic operations, that favor *numbers with relatively few blocks of contiguous* `0` *and* `1` *digits*, for which dramatic complexity reductions result even when operating on very large, "towers-of-exponents" numbers.

This paper is covers several other arithmetic operations on hereditarily binary numbers as well as their use for bitvector operations.

At the same time, we perform a performance evaluation confirming our theoretical complexity estimates of [1] and illustrate compact representations of some record-holder large prime numbers while emphasizing that hereditarily binary numbers replace bitsize with a measure of *structural complexity* as the key parameter deciding tractability of arithmetic operations.

[1] An extended draft version of [1] is available at at the *arxiv* repository [2].

The paper is organized as follows. Section 2 overviews basic definitions for *hereditarily binary numbers* and summarizes some of their properties, following [1]. Section 3 describes new arithmetic algorithms for hereditarily binary numbers. Section 4 illustrates the work of our algorithms on computations intractable with traditional bitstring representations. Section 5 compares the performance of several algorithms operating on hereditarily binary numbers with their conventional counterparts. Section 6 discusses related work. Section 7 concludes the paper and discusses future work. The Appendix wraps our arithmetic operations as instances of Haskell's number and order classes and illustrates compact tree-representations of some important number-theoretical entities like Mersenne, Fermat, Proth and Woodall primes.

We have adopted a *literate programming* style, i.e. the code contained in the paper forms a Haskell module (tested with ghc 7.6.3), available as a separate file at `http://logic.cse.unt.edu/tarau/research/2013/hbinx.hs`. It imports the code from [1], also available at `http://logic.cse.unt.edu/tarau/research/2013/hbin.hs`.

Alternatively, a Scala package implementing the same tree-based computations is available from `http://code.google.com/p/giant-numbers/`. We hope that this will encourage the reader to experiment interactively and validate the technical correctness of our claims.

2 Hereditarily Binary Numbers

We will summarize, following [1], the basic concepts behind *hereditarily binary numbers*. Through the paper we denote $\mathbb{N}$ the set of natural numbers and $\mathbb{N}^+$ the set of strictly positive natural numbers.

2.1 Bijective base-2 numbers

Natural numbers can be seen as represented by iterated applications of the functions $o(x) = 2x + 1$ and $i(x) = 2x + 2$ corresponding the so called *bijective base-2* representation [3], together with the convention that 0 is represented as the empty sequence. Each $n \in \mathbb{N}$ can be seen as a unique composition of these functions. We can make this precise as follows:

Definition 1 *We call bijective base-2 representation of $n \in \mathbb{N}$ the unique sequence of applications of functions o and i to ϵ that evaluates to n.*

With this representation, and denoting the empty sequence ϵ, one obtains $0 = \epsilon, 1 = o\ \epsilon, 2 = i\ \epsilon, 3 = o(o\ \epsilon), 4 = i(o\ \epsilon), 5 = o(i\ \epsilon)$ etc. The following

holds:

$$i(x) = o(x) + 1 \tag{1}$$

2.2 Efficient arithmetic with iterated functions o^n and i^n

Several arithmetic identities are proven in [1] and used to express efficient "one block of o^n or i^n operations at a time" algorithms for various arithmetic operations. Among them, we recall the following two, showing the connection with "left shift/multiplication by a power of 2" operations.

$$o^n(k) = 2^n(k+1) - 1 \tag{2}$$

$$i^n(k) = 2^n(k+2) - 2 \tag{3}$$

In particular

$$o^n(0) = 2^n - 1 \tag{4}$$

$$i^n(0) = 2^{n+1} - 2 \tag{5}$$

2.3 Hereditarily binary numbers as a data type

First we define a data type for our tree represented natural numbers, that we call *hereditarily binary numbers* to emphasize that *binary* rather than *unary* encoding is recursively used in their representation.

Definition 2 *The data type $\mathbb{T}$ of the set of hereditarily binary numbers is defined in [1] by the Haskell declaration:*

```
data T = E | V T [T]  | W T [T]
```

The intuition behind type $\mathbb{T}$ is the following:

- The term `E` (empty leaf) corresponds to zero
- the term `V x xs` counts the number `x+1` of `o` applications followed by an *alternation* of similar counts of `i` and `o` applications `xs`
- the term `W x xs` counts the number `x+1` of `i` applications followed by an *alternation* of similar counts of `o` and `i` applications `xs`

In [1] the bijection between $\mathbb{N}$ and $\mathbb{T}$ is provided by the function $n : \mathbb{T} \to \mathbb{N}$ and its inverse $t : \mathbb{N} \to \mathbb{T}$).

Definition 3 *The function $n : \mathbb{T} \to \mathbb{N}$ shown in equation* **6** *defines the unique natural number associated to a term of type $\mathbb{T}$.*

$$n(t) = \begin{cases} 0 & \text{if } t = \texttt{E}, \\ 2^{n(x)+1} - 1 & \text{if } t = \texttt{V x []}, \\ (n(u)+1)2^{n(x)+1} - 1 & \text{if } t = \texttt{V x (y:xs)} \text{ and } u = \texttt{W y xs}, \\ 2^{n(x)+2} - 2 & \text{if } t = \texttt{W x []}, \\ (n(u)+2)2^{n(x)+1} - 1 & \text{if } t = \texttt{W x (y:xs)} \text{ and } u = \texttt{V y xs}. \end{cases} \tag{6}$$

The following examples show the workings of the bijection n and illustrate that "structural complexity", defined in [1] as the *size of the tree representation without the root*, is bounded by the bitsize of a number and favors numbers in the neighborhood of towers of exponents of 2.

$2^{2^{16}} - 1 \rightarrow$ `V (V (V (V (V E[])[])[])[])[]` $\rightarrow 2^{2^{2^{2^{2^{0+1}-1+1}-1+1}-1+1}-1+1} - 1$

$20 \rightarrow$ `W E [E,E,E]` $\rightarrow (((2^{0+1} - 1 + 2)2^{0+1} - 2 + 1)2^{0+1} - 1 + 2)2^{0+1} - 2$

After defining constant average time *successor* and *predecessor* functions `s` and `s'`, constant average time definitions of of `o` and `i` are given in [1], as well as for the corresponding inverse operations that can be seen as "un-applying" a single instance of `o` or `i` and "recognizers" `e_` (corresponding to `E`), `o_` (corresponding to *odd* numbers) and `i_` (corresponding to *even* numbers).

3 Arithmetic operations working one o^k or i^k block at a time

[1] provides algorithms for basic arithmetic operations of addition (`sum`) subtraction (`sub`), exponent of 2 (`exp2`), multiplication by and division by a power of two (`leftshiftBy, rightshiftBy`) and arithmetic comparison (`cmp`). We will describe in this section algorithms for several other arithmetic operations, optimized for working *one block of o^k or i^k applications at a time*.

Reduced complexity general multiplication

Like in the case of `add` and `sub` we can derive a multiplication algorithm based on a several arithmetic identities involving exponents of 2 and iterated applications of the functions `o` and `i`.

Proposition 1 *The following holds:*

$$o^n(a)o^m(b) = o^{n+m}(ab + a + b) - o^n(a) - o^m(b) \tag{7}$$

Proof. By (2), we can expand and then reduce: $o^n(a)o^m(b) =$
$(2^n(a+1)-1)(2^m(b+1)-1) =$
$2^{n+m}(a+1)(b+1) - (2^n(a+1) + 2^m(b+1)) + 1 =$
$2^{n+m}(a+1)(b+1) - 1 - (2^n(a+1) - 1 + 2^m(b+1) - 1 + 2) + 2 =$
$o^{n+m}(ab+a+b+1) - (o^n(a) + o^m(b)) - 2 + 2 =$
$o^{n+m}(ab+a+b) - o^n(a) - o^m(b)$

Proposition 2

$$i^n(a)i^m(b) = i^{n+m}(ab + 2(a+b+1)) + 2 - i^{n+1}(a) - i^{m+1}(b) \qquad (8)$$

Proof. By (3), we can expand and then reduce: $i^n(a)i^m(b) =$
$(2^n(a+2)-2)(2^m(b+2)-2) =$
$2^{n+m}(a+2)(b+2) - (2^{n+1}(a+2) - 2 + 2^{m+1}(b+2) - 2) =$
$2^{n+m}(a+2)(b+2) - i^{n+1}(a) - i^{m+1}(b) =$
$2^{n+m}(a+2)(b+2) - 2 - (i^{n+1}(a) + i^{m+1}(b)) + 2 =$
$2^{n+m}(ab+2a+2b+2+2) - 2 - (i^{n+1}(a) + i^{m+1}(b)) + 2 =$
$i^{n+m}(ab+2a+2b+2) - (i^{n+1}(a) + i^{m+1}(b)) + 2 =$
$i^{n+m}(ab + 2(a+b+1)) + 2 - i^{n+1}(a) - i^{m+1}(b)$

The corresponding Haskell code starts with the obvious base cases:

```
mul _ E = E
mul E _ = E
```

When both terms represent odd numbers we apply the identity (7):

```
mul x  y | o_ x && o_ y = r2 where
  (n,a) = osplit x
  (m,b) = osplit y
  p1 = add (mul a b) (add a b)
  p2 = otimes (add (s n) (s m)) p1
  r1 = sub p2 x
  r2 = sub r1 y
```

The next two cases are reduced to the previous one by using the identity $i = s.o$.

```
mul x y | o_ x && i_ y = add x (mul x (s' y))
mul x y | i_ x && o_ y = add y (mul (s' x) y)
```

Finally, the most difficult case implements identity (8)

```
mul x y | i_ x && i_ y = r2 where
  (n,a) = isplit x
  (m,b) = isplit y
  p0 = mul a b
  s0 = s (add a b)
  p1 = add p0 (db s0)
```

```
  e1 = itimes (add (s n) (s m)) p1
  e2 = i x
  e3 = i y
  r1 = sub (s (s e1)) e2
  r2 = sub r1 e3
```

Note that when the operands are composed of large blocks of alternating o^n and i^m applications, the algorithm is quite efficient as it works (roughly) in time depending on the the number of blocks rather than the the number of digits. The following example illustrates a blend of arithmetic operations benefiting from complexity reductions on giant tree-represented numbers:

```
*HBinX> let term1 = sub (exp2 (exp2 (t 12345))) (exp2 (t 6789))
*HBinX> let term2 = add (exp2 (exp2 (t 123))) (exp2 (t 456789))
*HBinX> ilog2 (ilog2 (mul term1 term2))
V E [E,E,W E [],V E [E],E]
*HBinX> n it
12345
```

This example illustrates that our arithmetic operations are not limited by the size of their operands, but only by their "structural complexity".

3.1 Power

We first specialize our multiplication for a slightly faster squaring operation, using the identities:

$$(o^n(a))^2 = o^{2n}(a^2 + 2a) - 2o^n(a) \tag{9}$$

$$(i^n(a))^2 = i^{2n}(a^2 + 2(2a + 1)) + 2 - 2i^{n+1}(a) \tag{10}$$

```
square E = E
square x | o_ x = r where
  (n,a) = osplit x
  p1 = add (square a) (db a)
  p2 = otimes (i n) p1
  r = sub p2 (db x)
square x | i_ x = r where
  (n,a) = isplit x
  p1 = add (square a) (db (o a))
  p2 = itimes (i n) p1
  r = sub (s (s p2)) (db (i x))
```

We can implement a simple but efficient " power by squaring" operation x^y as follows:

```
pow _ E = V E []
pow x y | o_ y = mul x (pow (square x) (o' y))
pow x y | i_ y = mul x2 (pow x2 (i' y)) where x2 = square x
```

It works well with fairly large numbers, by also benefiting from efficiency of multiplication on terms with large blocks of o^n and o^m applications:

```
*HBinX> n (bitsize (pow (t 2014) (t 100)))
1097
*HBinX> pow (t 32) (t 10000000)
W E [W (W (V E []) []) [W E [E],V (V E []) [],E,E,E,W E [E],E]]
```

3.2 Division operations

A fairly efficient integer division algorithm is given here, but it does not provide the same complexity gains as, for instance, multiplication, addition or subtraction. Finding a "one block at a time" division algorithm, if possible at all, is subject of future work.

```
div_and_rem x y | LT == cmp x y = (E,x)
div_and_rem x y | y /= E = (q,r) where
  (qt,rm) = divstep x y
  (z,r) = div_and_rem rm y
  q = add (exp2 qt) z

  divstep n m = (q, sub n p) where
    q = try_to_double n m E
    p = leftshiftBy q m

  try_to_double x y k =
    if (LT==cmp x y) then s' k
    else try_to_double x (db y) (s k)

divide n m = fst (div_and_rem n m)
remainder n m = snd (div_and_rem n m)
```

3.3 Integer square root

A fairly efficient integer square root, using Newton's method, is implemented as follows:

```
isqrt E = E
isqrt n = if cmp (square k) n==GT then s' k else k where
  two = i E
  k=iter n
  iter x = if cmp (absdif r x)  two == LT
    then r
```

```
    else iter r where r = step x
  step x = divide (add x (divide n x)) two

absdif x y = if LT == cmp x y then sub y x else sub x y
```

3.4 Modular power

The modular power operation $x^y(mod\ m)$ can be optimized to avoid the creation of large intermediate results, by combining "power by squaring" and pushing the modulo operation inside the inner function `modStep`.

```
modPow m base expo = modStep expo (V E []) base where
  modStep (V E []) r b  = (mul r b) `remainder` m
  modStep x r b | o_ x =
    modStep (o' x) (remainder (mul r b) m) (remainder (square b)  m)
  modStep x r b = modStep (hf x) r (remainder (square b) m)
```

3.5 Lucas-Lehmer primality test for Mersenne numbers

The Lucas-Lehmer primality test has been used for the discovery of all the record holder largest known prime numbers oft the form $2^p - 1$ with p prime in the last few years. It is based on iterating $p - 2$ times the function $f(x) = x^2 - 2$, starting from $x = 4$. Then $2^p - 1$ is prime if and only if the result modulo $2^p - 1$ is 0, as proven in [4]. The function `ll_iter` implements this iteration.

```
ll_iter k n m | e_ k = n
ll_iter k n m = fastmod y m where
   x = ll_iter (s' k) n m
   y = s' (s' (square x))
```

It relies on the function `fastmod` which provides a specialized fast computation of $k\ modulo\ (2^p - 1)$.

```
fastmod k m | k == s' m = E
fastmod k m | LT == cmp k m = k
fastmod k m = fastmod (add q r) m where
  (q,r) = div_and_rem k m
```

Finally the Lucas-Lehmer test is implemented as follows:

```
lucas_lehmer (W E []) = True
lucas_lehmer p = e_ (ll_iter p_2 four m) where
  p_2 = s' (s' p)
  four = i (o E)
  m  = exp2 p
```

The following example illustrates its use for selecting a few Mersenne primes:

```
*HBinX> map n (filter lucas_lehmer (map t [3,5..31]))
[3,5,7,13,17,19,31]
*HBinX> map (\p->2^p-1) it
[7,31,127,8191,131071,524287,2147483647]
```

where the last line contains the Mersenne primes corresponding to p. We refer to the **Appendix** for the compact hereditarily binary representation of the largest known Mersenne prime.

3.6 Miller-Rabin probabilistic primality test

Let $\nu_2(x)$ denote the *dyadic valuation of* x, i.e., the largest exponent of 2 that divides x. The function `dyadicSplit` corresponding to

$$dyadicSplit(k) = (k, \frac{k}{2^{\nu_2(k)}}) \tag{11}$$

can be implemented as an average constant time operation as:

```
dyadicSplit z | o_ z = (E,z)
dyadicSplit z | i_ z = (s x, s (g xs)) where
  V x xs = s' z

  g [] =  E
  g (y:ys) = W y ys
```

After defining a sequence of `k` random natural numbers in an interval

```
randomNats seed k smallest largest = map t ns  where
  ns :: [Integer]
  ns = take k (randomRs (n smallest,n largest) (mkStdGen seed))
```

we are ready to implement the function `oddPrime` that runs `k` tests and concludes primality with probability $1 - \frac{1}{4^k}$ if all `k` calls to function `strongLiar` succeed.

```
oddPrime k m = all strongLiar as where
  m' = s' m
  as = randomNats k k (W E []) m'
  (l,d) = dyadicSplit m'

  strongLiar a = (x == V E []) || (any (== m') (squaringSteps l x)) where
    x = modPow m a d

    squaringSteps E _ = []
    squaringSteps l x = x:squaringSteps (s' l) (remainder (square x) m)
```

Note that we use `dyadicSplit` to find a pair `(l,d)` such that `l` is the largest power of `t 2` dividing the predecessor `m'` of the suspected prime

`m`. The function `strongLiar` checks for random base `a` the condition that primes (but possibly also a few composite numbers) verify.

Finally `isProbablyPrime` handles the case of even numbers. and calls `oddPrime` with the parameter specifying the number of tests, `k=42`.

```
isProbablyPrime (W E [])  = True
isProbablyPrime (W _ _) = False
isProbablyPrime p = oddPrime 42 p
```

The following example illustrates the correct behavior of the algorithm on a the interval `[2..100]`.

```
*HBinX> map n (filter isProbablyPrime (map t [2..100]))
[2,3,5,7,11,13,17,19,23,29,31,37,41,43,47,53,59,61,67,71,73,79,83,89,97]
```

4 Computing the Collatz/Syracuse sequence for huge numbers

An interesting application of our arithmetic algorithms, that achieves something one cannot do with ordinary Integers, is to explore computations on giant numbers, limited not by their sizes but by their structural complexity. The Collatz conjecture states that the function

$$collatz(x) = \begin{cases} \frac{x}{2} & \text{if } x \text{ is even,} \\ 3x+1 & \text{if } x \text{ is odd.} \end{cases} \tag{12}$$

reaches 1 after a finite number of iterations. An equivalent formulation, by grouping together all the division by 2 steps, is the function:

$$collatz'(x) = \begin{cases} \frac{x}{2^{\nu_2(x)}} & \text{if } x \text{ is even,} \\ 3x+1 & \text{if } x \text{ is odd.} \end{cases} \tag{13}$$

One more step further, by writing $k' = 2m+1$ we get a function that associates $k \in \mathbb{N}$ to $m \in \mathbb{N}$. One step further, the *syracuse function* is defined as the odd integer k' such that $n = 3k+1 = 2^{\nu(n)}k'$. The function `tl` computes efficiently the equivalent of

$$tl(k) = \frac{\frac{k}{2^{\nu_2(k)}} - 1}{2} \tag{14}$$

We can implement it by using the function `dyadicSplit` as follows:

```
tl k = o' (snd (dyadicSplit k))
```

Then our variant of the *syracuse function* corresponds to

$$syracuse(n) = tl(3n + 2) \tag{15}$$

which we can code efficiently as

```
syracuse n = tl (add n (i n))
```

The function `nsyr` computes the iterates of this function, until (possibly) stopping:

```
nsyr E = [E]
nsyr n = n : nsyr (syracuse n)
```

It is easy to see that the Collatz conjecture is true if and only if `nsyr` terminates for all n, as illustrated by the following example:

```
*HBinX> map n (nsyr (t 2014))
[2014,755,1133,1700,1275,1913,2870,1076,807,1211,1817,2726,1022,383,
 575,863,1295,1943,2915,4373,6560,4920,3690,86,32,24,18,3,5,8,6,2,0]
```

The next examples will show that computations for `nsyr` can be efficiently carried out for numbers that with traditional bitstring notations would easily overflow even the memory of a computer using as transistors all the atoms in the known universe. The following examples illustrate this:

```
map (n.tsize) (take 1000 (nsyr mersenne48))
[22,22,24,26,27,28,...,1292,1313,1335,1353]
```

As one can see, the structural complexity is growing progressively, but that our tree-numbers have no trouble with the computations.

```
*HBinX> map (n.tsize) (take 1000 (nsyr mersenne48))
[22,22,24,26,27,28,...,1292,1313,1335,1353]
```

Moreover, we can keep going up with a tower of 3 exponents. Interestingly, it results in a fairly small increase in structural complexity over the first 1000 terms.

```
*HBinX> map (n.tsize) (take 1000 (nsyr (exp2 (exp2 (exp2 mersenne48)))))
[26,33,36,37,40,42,...,1313,1335,1358,1375]
```

While we did not tried to wait out the termination of the execution for Mersenne number 48, we have computed `nsyr` for the record holder from 1952, which is still much larger than the values (up to 5×2^{60}) for which the conjecture has been confirmed true. Figure 1 shows the structural complexity curve for the "hailstone sequence" associated by the function `nsyr` to the 15-th Mersenne prime, $2^{1279} - 1$

As an interesting fact, possibly unknown so far, one might notice the abrupt phase transition that, based on our experiments, seem to characterize the behavior of this function, when starting with very large numbers of relatively small structural complexity.

And finally something we are quite sure has never been computed before, we can also start with a *tower of exponents 100 levels tall*:



Fig. 1: Structural complexity curve for `nsyr` on $2^{1279} - 1$

```
*HBinX> take 1000 (map(n.tsize)(nsyr (bestCase (t 100))))
[100,296,298,300,301,302,...,1587,1576,1597,1619,1637]
```

5 Performance evaluation

In [1] functions returning worst and best case tree representations are defined. The worst case representation of hereditarily binary numbers is proportional to their bitsize:

```
> worstCase 5
W E [E,E,E,E,E,E,E,E]
> n it
1364
```

On the other hand, the "tower of exponents" best case provides a very compact representation to gigantic numbers:

```
> bestCase 5
W (W (W (W (W E []) []) []) []) []
> n (bitsize (bitsize it))
65536
```

Therefore, it is clearly possible, provided that the algorithms in [1] are as efficient as claimed, to have computations with such giant numbers that are intractable even with the very best arbitrary integer packages like the GMP library used by Haskell.

At the same time, GMP is written in highly optimized compiled C (+ some assembler), while our hereditarily binary numbers are supported by a tree data structure in interpreted Haskell (ghci). Consequently, one should expect large, up to 2-3 orders of magnitude constant factors on computations where our worst case dominates. On the other hand, the our

constant time successor and predecessor algorithm will, surprisingly, be very close to its native counterpart, while addition/subtraction-intensive operations will be somewhere in-between.

Our performance measurements (run on a Mac Air with 8GB of memory and an Intel i7 processor) serve two objectives:

1. to show that, on the average, our tree based representations perform on a blend of arithmetic computations within a constant factor compared with conventional bitstring-based computations
2. to show that on interesting special computations they match or outperform the bitstring-based arbitrary size `Integer`, due to the much lower asymptotic complexity of such operations on data type $\mathbb{T}$, despite of the 2-3 orders of magnitude gap with GMP.

Finally, let us emphasize that this comparison is by no means a benchmarking of two competing arbitrary integer arithmetic libraries, but rather a straightforward empirical check that our arithmetic algorithms have the complexity predicted by their theoretical study in [1] and that they work within their expected complexity bounds.

for instance, objective *1* is well served by the Ackerman function that exercises the successor and predecessor functions quite heavily.

On the other hand, objective *2* is served by benchmarks that take advantage of the compressed representation of very large, record holder primes. In some cases, the conventional representations are unable to run these benchmarks within existing computer memory and CPU-power limitations, as marked with "`?`" in the comparison table of figure 2. In other cases, like in the "`v (t 22) (t 11)`" benchmark, computing a very large boolean projection variable made of relatively few alternating blocks of `0`s and `1`s, data type $\mathbb{T}$ performs significantly faster than binary representations. The last two rows in figure 2 show two more extreme examples. The first computes `1000` steps of the *Syracuse function* (an equivalent of the em Collatz conjecture), starting from a giant towers of exponents number and the second multiplies together 5 very large, record holder prime numbers.

Together they indicate that optimized libraries derived from our tree-based representations are likely to be competitive with existing bitstring-based packages on typical computations and outperform them by an arbitrary margin on some number-theoretically interesting computations. While the code of the benchmarks is omitted due to space constraints, it is part of the companion Haskell file at `http://logic.cse.unt.edu/tarau/Research/2013/hbinx.hs`.

Benchmark	Integer	tree type $\mathbb{T}$
$2^{2^{30}}$	10192	0
v 22 11	4850	297
Ackermann 3 7	491	718
2^{21} predecessors	1979	2330
fibonacci 30	3249	19414
sum of first 2^{16} naturals	68	10016
powers	46	13485
generating primes	6	4807
factorial of 200	2	8040
1000 syracuse steps from $2^{\cdots^{2^2}}$	?	9070
product of 5 giant primes	?	904

Fig. 2: Time (in ms.) on a few small benchmarks

6 Related work

This paper is a sequel to [1] where hereditary binary numbers are introduced and "one block of iterated o and i operations at a time" algorithms are described, together with the number-theoretical identities they are relying on.

As an extension of [1], where basic arithmetic algorithms that take advantage of our tree based number representation like multiplication, addition and comparison are described, this paper covers several other arithmetic algorithms as well as bitvector boolean logic and an empirical validation of various performance bounds.

Several notations for very large numbers have been invented in the past. Examples include Knuth's *arrow-up* notation [5] covering operations like the *tetration* (a notation for towers of exponents). In contrast to our tree-based natural numbers, such notations are not closed under addition and multiplication, and consequently they cannot be used as a replacement for ordinary binary or decimal numbers.

The first instance of a hereditary number system, at our best knowledge, occurs in the proof of Goodstein's theorem [6], where replacement of finite numbers on a tree's branches by the ordinal ω allows him to prove that a "hailstone sequence" visiting arbitrarily large numbers eventually turns around and terminates.

Numeration systems on regular languages have been studied recently, e.g. in [7] and specific instances of them are also known as bijective base-

k numbers. Arithmetic packages similar to our bijective base-2 view of arithmetic operations are part of libraries of proof assistants like Coq [8].

In [9] integer decision diagrams are introduced providing a compressed representation for sparse integers, sets and various other data types.

7 Conclusion and future work

We have validated the complexity bounds of our arithmetic algorithms on hereditarily binary numbers introduced in [1], defined several new arithmetic algorithms for them and applied our tree-base number representations to bitvector logic.

Our performance analysis has shown the wide spectrum of best and worst case behaviors of our arithmetic algorithms when compared to Haskell's GMP-based `Integer` operations.

As planned in [1], future work will focus on developing a practical arithmetic library based on a hybrid representation, where the leaves the empty leaves of our trees will be replaced with 64-bit integers, to benefit from fast hardware arithmetic on small numbers.

References

1. Tarau, P., Buckles, B.: Arithmetic Algorithms for Hereditarily Binary Natural Numbers. In: Proceedings of SAC'14, ACM Symposium on Applied Computing, PL track, Gyeongju, Korea, ACM (March 2014) to appear.
2. Tarau, P.: Arithmetic Algorithms for Hereditarily Binary Natural Numbers (June 2013) http://arxiv.org/abs/1306.1128.
3. Wikipedia: Bijective numeration — wikipedia, the free encyclopedia (2012) [Online; accessed 2-June-2012].
4. Bruce, J.W.: A Really Trivial Proof of the Lucas-Lehmer Test. The American Mathematical Monthly **100**(4) (1993) 370–371
5. Knuth, D.E.: Mathematics and Computer Science: Coping with Finiteness. Science **194**(4271) (1976) 1235 –1242
6. Goodstein, R.: On the restricted ordinal theorem. Journal of Symbolic Logic (9) (1944) 33–41
7. Rigo, M.: Numeration systems on a regular language: arithmetic operations, recognizability and formal power series. Theoretical Computer Science **269**(1-2) (2001) 469 – 498
8. The Coq development team: The Coq proof assistant reference manual. LogiCal Project. (2012) Version 8.4.
9. Vuillemin, J.: Efficient Data Structure and Algorithms for Sparse Integers, Sets and Predicates. In: Computer Arithmetic, 2009. ARITH 2009. 19th IEEE Symposium on. (June 2009) 7 –14
10. : Great Internet Mersenne Prime Search (2013)

Appendix

Implementing Haskell's Number and Order-related Type Classes

In [1] algorithms working "one block of o^n or i^m applications at a time" are introduced for various arithmetic operations on type $\mathbb{T}$, and implemented as the following functions:

- `add` : addition
- `sub` : subtraction
- `mul` : multiplication
- `divide` : integer division
- `remainder` : remainder of integer division
- `cmp`: comparison operation, returning `LT,EQ,GT`
- `exp2`: exponent of 2
- `leftshiftBy x y`: specialized multiplication $2^x y$
- `rightshiftBy x y`: specialized integer division $\frac{y}{2^x}$
- `pow`: power operation
- `bitsize`: computing the bitsize of a bijective base-2 representation
- `tsize`: computing the structural complexity of a tree-represented number
- `dual`: flipping of topmost `V` and `W` constructors, corresponding to flipping of all `o` and `i` applications.

We will make here the type $\mathbb{T}$ an instance of Haskell's predefined number and order-related type classes like `Ord` and `Num`, etc., to enable the usual arithmetic operation and comparison syntax.

```
instance Ord T where compare = cmp
instance Num T where
  (+) = add
  (-) = sub
  (*) = mul
  fromInteger = t
  abs = id
  signum E = E
  signum _ = V E []
instance Integral T where
  quot = divide
  div = divide
  rem = remainder
  mod = remainder
  quotRem = div_and_rem
  divMod = div_and_rem
  toInteger = n
```

```
instance Real T where
  toRational = toRational . n
instance Enum T where
  fromEnum = fromEnum . n
  toEnum = t . f where
    f:: Int→Integer
    f = toEnum
  succ = s
  pred = s'
```

Note that as Haskell does not have a built-in natural number class, we mapped `abs` and `signum` to identity `id` and the constant value corresponding to 1, `V E []`.

The following example illustrates the use of the `+` operation directly on objects of type $\mathbb{T}$:

```
> t 3
V (V E []) []
> t 4
W E [E]
> (V (V E []) []) + (W E [E])
V (W E []) []
> n it
7
```

Compact representation of some record-holder giant numbers

Let's first observe that Fermat, Mersenne, perfect numbers have all compact expressions with our tree representation of type $\mathbb{T}$.

```
fermat n = s (exp2 (exp2 n))

mersenne p = s' (exp2 p)

perfect p = s (V q [q]) where q = s' (s' p)
```

```
HBin> mersenne 127
170141183460469231731687303715884105727
> mersenne (t 127)
V (W (V E [E]) []) []
```

The largest known prime number, found by the GIMPS distributed computing project [10] in January 2013 is the 48-th Mersenne prime $= 2^{57885161} - 1$ (with possibly smaller Mersenne primes below it). It is defined in Haskell as follows:

```
-- exponent of the 48-th known Mersenne prime
prime48 = 57885161
-- the actual Mersenne prime
mersenne48 = s' (exp2 (t prime48))
```

While it has a bit-size of 57885161, its compressed tree representation is rather small:

```
> mersenne48
V (W E [V E [],E,E,V (V E []) [],
   W E [E],E,E,V E [],V E [],W E [],E,E]) []
```

The equivalent DAG representation of the 48-th Mersenne prime, shown in Figure 3, has only 7 shared nodes and structural complexity 22.



Fig. 3: Largest known prime number discovered in January 2013: the 48-th Mersenne prime, represented as a DAG

It is interesting to note that similar compact representations can also be derived for perfect numbers. For instance, the largest known perfect number, derived from the largest known Mersenne prime as $2^{57885161-1}(2^{57885161}-1)$, (involving only 8 shared nodes and structural complexity 43) is:

```
perfect48 = perfect (t prime48)
```

Fig. 4 shows the DAG representation of the largest known perfect number, derived from Mersenne number 48. Similarly, the largest Fermat number



Fig. 4: Largest known perfect number in January 2013

that has been factored so far, $F_{11} = 2^{2^{11}} + 1$ is compactly represented as

```
> fermat (t 11)
V E [E,V E [W E [V E []]]]
```

with structural complexity 8. By contrast, its (bijective base-2) binary representation consists of 2048 digits.

Moreover, even "towering up" `mersenne` numbers also leads to compact representations. For instance the Catalan sequence is defined as:

```
catalan E = i E
catalan n = s' (exp2 (catalan (s' n)))
```

and Catalan's conjecture states that they are all primes. The conjecture is still unsolved, and as the following example shows, even primality of `catalan 5` is currently intractable.

```
> catalan (t 5)
V (W (V E [W E [E]]) []) []
> n (tsize (catalan (t 5)))
6
> n (bitsize (catalan (t 5)))
170141183460469231731687303715884105727
```

Figures 5 and 6 show the DAG representations of Catalan-Mersenne numbers 5 and 10.



Fig. 5: Catalan-Mersenne number 10

Some other very large primes that are not Mersenne numbers also have compact representations.

The generalized Fermat prime $27653 * 2^{9167433} + 1$, (currently the 15-the largest prime number) computed as a tree (with 7 shared nodes and structural complexity 30) is:

```
genFermatPrime = s (leftshiftBy n k) where
  n = t (9167433::Integer)
  k = t (27653::Integer)
```



Fig. 6: Catalan-Mersenne number 10

```
> genFermatPrime
V E [E,W (W E []) [W E [],E,
  V E [],E,W E [],W E [E],E,E,W E []],
  E,E,E,W (V E []) [],V E [],E,E]
```

Figure 7 shows the DAG representation of this generalized Fermat prime.



Fig. 7: Generalized Fermat prime

The largest known Cullen prime $6679881 * 2^{6679881} + 1$ computed as tree (with 6 shared nodes and structural complexity 43) is:

```
cullenPrime = s (leftshiftBy n n) where
  n = t (6679881::Integer)
```

```
> cullenPrime
V E [E,W (W E []) [W E [],E,E,E,E,V E [],E,
  V (V E []) [],E,E,V E [],E],E,V E [],E,V E [],
  E,E,E,E,V E [],E,V (V E []) [],E,E,V E [],E]
```

Figure 8 shows the DAG representation of this Cullen prime. The largest



Fig. 8: largest known Cullen prime

known Woodall prime $3752948 * 2^{3752948} - 1$ computed as a tree (with 6 shared nodes and structural complexity 33) is:

```
woodallPrime = s' (leftshiftBy n n) where
  n = t (3752948::Integer)
```

```
> woodallPrime
V (V E [V E [],E,V E [E],V (V E []) [],
 E,E,E,V E [],V E []]) [E,E,V E [E],
 V (V E []) [],E,E,E,V E [],V E []]
```

Figure 9 shows the DAG representation of this Woodall prime.



Fig. 9: Largest known Woodall prime

The largest known Proth prime $19249 * 2^{13018586} + 1$ computed as a tree is:

```
prothPrime = s (leftshiftBy n k) where
  n = t (13018586::Integer)
  k = t (19249::Integer)
```

```
> prothPrime
V E [E,V (W E []) [V E [],E,W E [],E,E,
  V E [],E,E,E,E,V E [],W E [],E],E,W E [],
  V E [],V E [],V E [],E,E,V E []]
```

Figure 10 shows the DAG representation of this Proth prime, the largest non-Mersenne prime known by March 2013 with 5 shared nodes and structural complexity 36.



Fig. 10: Largest known Proth prime

The largest known Sophie Germain prime $18543637900515 * 2^{666667} - 1$ computed as a tree (with 6 shared nodes and structural complexity 56) is:

```
sophieGermainPrime = s' (leftshiftBy n k) where
  n = t (666667::Integer)
  k = t (18543637900515::Integer)
```

```
> sophieGermainPrime
V (W (V E []) [E,E,E,E,V (V E []) [],V E [],E,
   E,W E [],E,E]) [V E [],W E [],W E [],V E [],
   V E [],E,E,V E [],V E [],V E [],V (V E [])
   [],E,V E [],V (V E []) [],V E [],E,W E [],E,
   V E [],V (V E []) []]
```

Figure 11 shows the DAG representation of this prime. The largest known twin primes $3756801695685 * 2^{666669} \pm 1$ computed as a pair of trees (with 7 shared nodes both and structural complexities of 54 and 56) are:

```
twinPrimes = (s' m,s m) where
  n = t (666669::Integer)
  k = t (3756801695685::Integer)
  m = leftshiftBy n k
```



Fig. 11: Largest known Sophie Germain prime

```
> fst twinPrimes
V (W E [E,V E [],E,E,V (V E []) [],
   V E [],E,E,W E [],E,E])
   [E,E,E,W E [],W (V E []) [],V E [],E,V E [],
   E,E,E,E,V E [],E,E,
   V E [],V E [],E,E,E,E,E,E,E,V E [],E,E]
> snd twinPrimes
V E [E,W (V E []) [E,E,E,E,V (V E []) [],
  V E [],E,E,W E [],E,E],E,E,E,
  W E [],W (V E []) [],V E [],E,V E [],
  E,E,E,E,V E [],E,E,V E [],
  V E [],E,E,E,E,E,E,E,V E [],E,E]
```

Figures 12 and 13 show the DAG representation of these twin primes.



Fig. 12: Largest known twin prime 1



Fig. 13: Largest known twin prime 2

One can appreciate the succinctness of our representations, given that all these numbers have hundreds of thousands or millions of decimal digits. An interesting challenge would be to (re)focus on discovering primes with significantly larger structural complexity then the current record holders by bitsize.

More importantly, as the following examples illustrate it, *computations* like addition, subtraction and multiplication of such numbers are possible:

```
> sub genFermatPrime (t 2014)
V (V E []) [E,V E [],E,W E [E],V E [W E [E],
  W E [],E,W E [],W E [E],E,E,W E []],V E [],
  E,W (V E []) [],V E [],E,E]
> bitsize (sub prothPrime (t 1234567890))
W E [V E [],E,E,V (V E []) [],E,E,
  V E [],E,E,E,E,V E [],W E [],E]
> tsize (exp2 (exp2 mersenne48))
V E [E,E,E]
> tsize(leftshiftBy mersenne48 mersenne48)
V E [W E [],E]
> add (t 2) (fst twinPrimes) ==
           (snd twinPrimes)
True
> ilog2 (ilog2 (mul prothPrime cullenPrime))
W E [V E [],E]
> n it
24
```