

Declarative Combinatorics: on Structural Isomorphisms between Pairing/Unpairing Trees and Binary Decision Diagrams

Paul Tarau

Department of Computer Science and Engineering
University of North Texas
E-mail: tarau@cs.unt.edu

Abstract. A *pairing function* J associates a unique natural number z to any two natural numbers x, y such that for two *unpairing functions* K and L , the equalities $K(J(x, y)) = x$, $L(J(x, y)) = y$ and $J(K(z), L(z)) = z$ hold. Using pairing functions on natural number representations of truth tables, we derive an encoding for Binary Decision Diagrams with the unique property that its boolean evaluation faithfully mimics its structural conversion to a natural number through recursive application of a matching pairing function. We then use this result to derive *ranking* and *unranking* functions for BDDs and reduced BDDs with application to uniform generation of random BDDs. The paper is organized as a self-contained literate Prolog program, available at <http://logic.csci.unt.edu/tarau/research/2009/pBDD.zip>.

Keywords: logic programming and computational mathematics, pairing/unpairing functions, encodings of boolean functions, binary decision diagrams, natural number representations of truth tables

1 Introduction

This paper is an exploration with logic programming tools of *ranking* and *unranking* problems on Binary Decision Diagrams. The practical expressiveness of logic programming languages (in particular Prolog) are put at test in the process. The paper is part of a larger effort to cover in a declarative programming paradigm some fundamental combinatorial generation algorithms. However, our main focus is by no means “yet another implementation of BDDs in Prolog”. The paper is more about fundamental isomorphisms between logic functions and their natural number representations, in the tradition of [1], with the unusual twist that everything is expressed as a literate Prolog program, and therefore automatically testable by the reader. One could put such efforts under the generic umbrella of an emerging research field that we would like to call *executable theoretical computer science*. Nevertheless, we also hope that the more practically oriented reader will be able to benefit from this approach by being able to experiment with, and reuse our Prolog code in applications.

The paper is organized as follows: Sections 2 and 3 overview efficient evaluation of boolean formulae in Prolog using bitvectors represented as arbitrary length integers and Binary Decision Diagrams (BDDs).

Section 4 discusses classic pairing and unpairing operations and introduces pairing/unpairing predicates acting directly on bitlists.

Section 5 introduces a novel BDD encoding (based on our unpairing functions) and discusses the surprising equivalence between boolean evaluation of BDDs and the inverse of our encoding, the main result of the paper.

Section 6 describes *ranking* and *unranking* functions for BDDs and reduced BDDs.

Sections 8 and 9 discuss related work, future work and conclusions.

The code in the paper, embedded in a literate programming LaTeX file, is entirely self contained and has been tested under *SWI-Prolog*.

2 Parallel Evaluation of Boolean Functions with Bitvector Operations

Evaluation of a boolean function can be performed one value at a time as in the predicate `if_then_else/4`

```
if_then_else(X,Y,Z,R):-bit(X),bit(Y),bit(Z),
    ( X==1->R=Y
    ; R=Z
    ).
```

```
bit(0).
bit(1).
```

resulting in a *truth table*¹

```
?- if_then_else(X,Y,Z,R),write([X,Y,Z]:R),nl,fail;nl.
[0, 0, 0]:0
[0, 0, 1]:1
[0, 1, 0]:0
[0, 1, 1]:1
[1, 0, 0]:0
[1, 0, 1]:0
[1, 1, 0]:1
[1, 1, 1]:1
```

Clearly, this does not take advantage of the ability of modern hardware to perform such operations one word a time - with the instant benefit of a speed-up proportional to the word size. An alternate representation, adapted from [2] uses integer encodings of 2^n bits for each boolean variable X_0, \dots, X_{n-1} . Bitvector operations evaluate all value combinations at once.

¹ One can see that if the number of variables is fixed, we can ignore the bitstrings in the brackets. Thus, the truth table can be identified with the natural number represented in binary form by the last column.

Proposition 1 *Let x_k be a variable for $0 \leq k < n$ where n is the number of distinct variables in a boolean expression. Then column k in the matrix representation of the inputs in the truth table represents, as a bitstring, the natural number:*

$$x_k = (2^{2^n} - 1) / (2^{2^{n-k-1}} + 1) \quad (1)$$

For instance, if $n = 2$, the formula computes $x_0 = 3 = [0, 0, 1, 1]$ and $x_1 = 5 = [0, 1, 0, 1]$.

The following predicates, working with arbitrary length bitstrings are used to evaluate variables x_k with $k \in [0..n-1]$ with formula 1 and map the constant boolean function 1 to the bitstring of length 2^n , 111...1, representing $2^{2^n} - 1$

```
% maps variable K in [0..NbOfBits-1] to Xk
var_to_bitstring_int(NbOfBits,K,Xk):-
    all_ones_mask(NbOfBits,Mask),
    var_to_bitstring_int(NbOfBits,Mask,K,Xk).

var_to_bitstring_int(NbOfBits,Mask,K,Xk):-
    NK is NbOfBits-(K+1),
    D is (1<<<(1<<NK))+1,
    Xk is Mask//D.

% represents constant 1 as 11...1 build with NbOfBits bits
all_ones_mask(NbOfBits,Mask):-Mask is (1<<<(1<<<NbOfBits))-1.
```

We have used in `var_to_bitstring_int` an adaptation of the efficient bitstring-integer encoding described in the Boolean Evaluation section of [2]. Intuitively, it is based on the idea that one can look at n variables as bitstring representations of the n columns of the truth table.

Variables representing such bitstring-truth tables (seen as *projection functions*) can be combined with the usual bitwise integer operators, to obtain new bitstring truth tables, encoding all possible value combinations of their arguments. Note that the constant 0 is represented as 0 while the constant 1 is represented as $2^{2^n} - 1$, corresponding to a column in the truth table containing ones exclusively.

3 Binary Decision Diagrams

We have seen that Natural Numbers in $[0..2^{2^n} - 1]$ can be used as representations of truth tables defining n -variable boolean functions. A binary decision diagram (BDD) [3] is an ordered binary tree obtained from a boolean function, by assigning its variables, one at a time, to 0 (left branch) and 1 (right branch). In virtually all practical applications BDDs are represented as DAGs after detecting shared nodes. We safely ignore this here as they represent the same logic function, which is all we care about at this point. Typically in the early literature, the acronym ROBDD is used to denote reduced ordered BDDs. Because

this optimization is now so prevalent, the term BDD is frequently use to refer to ROBDDs. Strictly speaking, BDD in this paper will stand for *ordered BDD with reduction of identical branches but without node sharing*.

The construction deriving a BDD of a boolean function f is known as Shannon expansion [4], and is expressed as

$$f(x) = (\bar{x} \wedge f[x \leftarrow 0]) \vee (x \wedge f[x \leftarrow 1]) \quad (2)$$

where $f[x \leftarrow a]$ is computed by uniformly substituting a for x in f . Note that by using the more familiar boolean if-the-else function Shannon expansion can also be expressed as:

$$f(x) = \text{if } x \text{ then } f[x \leftarrow 1] \text{ else } f[x \leftarrow 0] \quad (3)$$

We represent a *BDD* in Prolog as a binary tree with constants 0 and 1 as leaves, marked with the function symbol `c/1`. Internal *if-then-else* nodes marked with `ite/3` are controlled by variables, ordered identically in each branch, as first arguments of `ite/1`. The two other arguments are subtrees representing the **Then** and **Else** branches. Note that, in practice, reduced, canonical DAG representations are used instead of binary tree representations.

Alternatively, we observe that the Shannon expansion can be directly derived from a 2^n size truth table, using bitstring operations on encodings of its n variables. Assuming that the first column of a truth table corresponds to variable x , $x = 0$ and $x = 1$ mask out, respectively, the upper and lower half of the truth table. The predicate `shannon_split/4` splits a truth table of NV variables in 2 tables of $NV-1$ variables.

```
shannon_split(NV,X, Hi,Lo):-
    all_ones_mask(NV,M),NV1 is NV-1,
    all_ones_mask(NV1,LM),
    HM is xor(M,LM),
    Lo is /\(LM,X),H is /\(HM,X),
    Hi is H>>(1<<NV1).
```

Note that the operation `shannon_split` can be reversed by fusing 2 truth tables of $NV-1$ variables into one of NV variables:

```
shannon_fuse(NV,Hi,Lo, X):-
    NV1 is NV-1,
    H is Hi<<(1<<NV1),
    X is \/ (H,Lo).
```

```
?- shannon_split(2, 7, X,Y),shannon_fuse(2, X,Y, Z).
X = 1,
Y = 3,
Z = 7.
```

```
?- shannon_split(3, 42, X,Y),shannon_fuse(3, X,Y, Z).
X = 2,
Y = 10,
Z = 42.
```

Another way to look at these two operations (for a fixed value of NV), is as bijections associating a pair of natural numbers to a natural number, i.e. as *pairing* functions.

4 Pairing and Unpairing Functions

Definition 1 A pairing function is a bijection $f : Nat \times Nat \rightarrow Nat$. An unpairing function is a bijection $g : Nat \rightarrow Nat \times Nat$.

Following Julia Robinson's notation [5], given a pairing function J , its left and right inverses K and L are such that

$$J(K(z), L(z)) = z \quad (4)$$

$$K(J(x, y)) = x \quad (5)$$

$$L(J(x, y)) = y \quad (6)$$

We refer to [6] for a typical use in the foundations of mathematics and to [7] for an extensive study of various pairing functions and their computational properties.

4.1 Cantor's Pairing Function

Starting from Cantor's pairing function

```
cantor_pair(K1,K2,P):-P is (((K1+K2)*(K1+K2+1))/2)+K2.
```

bijections from $Nat \times Nat$ to Nat have been used for various proofs and constructions of mathematical objects [5, 6].

For $X, Y \in \{0, 1, 2, 3\}$ the sequence of values of this pairing function is:

```
?- findall(R, (between(0,3,A), between(0,3,B), cantor_pair(A,B,R)), Rs).
Rs = [0, 2, 4, 6, 1, 5, 9, 13, 3, 11, 19, 27, 7, 23, 39, 55]
```

Note however, that the inverse of Cantor's pairing function involves floating point operations that require emulation in terms of arbitrary length integers to avoid loosing precision.

4.2 Pairing/Unpairing operations acting directly on bitlists

We will describe here pairing operations, that are expressed exclusively as bitlist transformations of `bitunpair` and its inverse `bitpair`, and are therefore likely to be easily hardware implementable. As we have found out recently, they turn out to be the same as the functions defined in Steven Pigeon's PhD thesis on Data Compression [8], page 114).

The predicate `bitpair` implements a bijection from $Nat \times Nat$ to Nat that works by splitting a number's big endian bitstring representation into odd and

even bits, while its inverse `to_pair` blends the odd and even bits back together. The helper predicates `to_rbits` and `from_rbits`, given in the Appendix, convert to/from integers to bitlists.

```
bitpair(X,Y,P):-
    to_rbits(X,Xs),
    to_rbits(Y,Ys),
    bitmix(Xs,Ys,Ps),!,
    from_rbits(Ps,P).

bitunpair(P,X,Y):-
    to_rbits(P,Ps),
    bitmix(Xs,Ys,Ps),!,
    from_rbits(Xs,X),
    from_rbits(Ys,Y).

bitmix([X|Xs],Ys,[X|Ms]):-!,bitmix(Ys,Xs,Ms).
bitmix([], [X|Xs], [0|Ms]):-!,bitmix([X|Xs], [], Ms).
bitmix([], [], []).
```

The transformation of the bitlists, done by the bidirectional predicate `bitmix/2` is shown in the following example with bitstrings aligned:

```
?- bitunpair(2008,X,Y),bitpair(X,Y,Z).
X = 60,
Y = 26,
Z = 2008

% 2008:[0, 0, 0, 1, 1, 0, 1, 1, 1, 1, 1]
% 60:[0, 0, 1, 1, 1, 1]
% 26:[ 0, 1, 0, 1, 1]
```

Note that we represent numbers with bits in reverse order (least significant first). Like in the case of Cantor's pairing function, we can see similar growth in both arguments:

```
?- between(0,15,N),bitunpair(N,A,B),
    write(N:(A,B)),write(' '),fail;nl.
0: (0, 0) 1: (1, 0) 2: (0, 1) 3: (1, 1)
4: (2, 0) 5: (3, 0) 6: (2, 1) 7: (3, 1)
8: (0, 2) 9: (1, 2) 10: (0, 3) 11: (1, 3)
12: (2, 2) 13: (3, 2) 14: (2, 3) 15: (3, 3)

?- between(0,3,A),between(0,3,B),bitpair(A,B,N),
    write(N:(A,B)),write(' '),fail;nl.
0: (0, 0) 2: (0, 1) 8: (0, 2) 10: (0, 3)
1: (1, 0) 3: (1, 1) 9: (1, 2) 11: (1, 3)
4: (2, 0) 6: (2, 1) 12: (2, 2) 14: (2, 3)
5: (3, 0) 7: (3, 1) 13: (3, 2) 15: (3, 3)
```

It is also convenient sometimes to see pairing/unpairing as one-to-one functions from/to the underlying language's ordered pairs, i.e. `X-Y` in Prolog :

```

bitpair(X-Y,Z):-bitpair(X,Y,Z).

bitunpair(Z,X-Y):-bitunpair(Z,X,Y).

```

5 Encodings of Binary Decision Diagrams as Pairing/Unpairing Trees

We will build a *pairing/unpairing tree* by applying `bitunpair` recursively to a Natural Number `TT`, seen as an N -variable 2^N bit truth table. This results in a complete binary tree of depth N . As we will show later, this binary tree represents an unreduced *BDD* that returns `TT` when evaluated applying its boolean operations. The predicate `unPairingTree/3` constructs the *pairing/unpairing tree*. Its input arguments are `NV` counting the number of variables, `TT` an integer denoting a truth table. In anticipation of the equivalence with BDDs shown in subsection 5.4 we will use the functor `bdd/2` to designate the output term built with variables `NV` and `unPairingTree`.

```

unPairingTree(NV,TT, bdd(NV,UnPairingTree)):-
    Max is (1<<<(1<<NV)), TT<Max,
    isplit(NV,TT, UnPairingTree).

isplit(0,TT,c(TT)).
isplit(NV,TT,ite(NV1,H,L)):-NV>0,
    NV1 is NV-1,
    bitunpair(TT,Hi,Lo),
    isplit(NV1,Hi,H),
    isplit(NV1,Lo,L).

```

The actual work is accomplished by `isplit/3` the recurses to depth `NV`, splitting `TT` into pairs. The following examples show the results returned by `unPairingTree/3` for all 2^{2^k} truth tables associated to k variables, with $k = 2$.

```

?- between(0,15,TT),unPairingTree(2,TT,BDD),write(TT:BDD),nl,fail,nl
0:bdd(2, ite(1, ite(0, c(0), c(0)), ite(0, c(0), c(0))))
1:bdd(2, ite(1, ite(0, c(1), c(0)), ite(0, c(0), c(0))))
2:bdd(2, ite(1, ite(0, c(0), c(0)), ite(0, c(1), c(0))))
...
13:bdd(2, ite(1, ite(0, c(1), c(1)), ite(0, c(0), c(1))))
14:bdd(2, ite(1, ite(0, c(0), c(1)), ite(0, c(1), c(1))))
15:bdd(2, ite(1, ite(0, c(1), c(1)), ite(0, c(1), c(1))))

```

Note that we have named the nodes `ite/3` in anticipation of the result in subsection 5.4 where they are shown to represent if-then-else nodes in a *BDD*.

5.1 Reducing the Pairing/Unpairing Trees

The predicate `uTreeReduce` reduces a *BDD* by trimming identical left and right subtrees, and the predicate `reducedUnpairingTree` associates this reduced form to the truth table `TT` represented as a natural number.

```
uTreeReduce(BDD,bdd(NV,R)):-nonvar(BDD),BDD=bdd(NV,X),uTreeReduce1(X,R).
```

```
uTreeReduce1(c(TT),c(TT)).
uTreeReduce1(ite(_,A,B),R):-A=B,uTreeReduce1(A,R).
uTreeReduce1(ite(X,A,B),ite(X,RA,RB)):-A\=B,
    uTreeReduce1(A,RA),uTreeReduce1(B,RB).
```

```
reducedUnpairingTree(NV,TT, ReducedBDD):-
    unPairingTree(NV,TT, BDD),
    uTreeReduce(BDD,ReducedBDD).
```

Note that we omit here the reduction step consisting in sharing common subtrees, as it is obtained easily by replacing trees with DAGs. The process is facilitated by the fact that our unique encoding provides a perfect hashing key for each subtree. The following examples show the results returned by `reducedUnpairingTree` for `NV=2`.

```
?- between(0,15,TT),reducedUnpairingTree(2,TT,BDD),write(TT:BDD),nl,fail;nl.
0:bdd(2, c(0))
1:bdd(2, ite(1, ite(0, c(1), c(0)), c(0)))
2:bdd(2, ite(1, c(0), ite(0, c(1), c(0))))
3:bdd(2, ite(0, c(1), c(0)))
...
13:bdd(2, ite(1, c(1), ite(0, c(0), c(1))))
14:bdd(2, ite(1, ite(0, c(0), c(1)), c(1)))
15:bdd(2, c(1))
```

5.2 From BDDs to Natural Numbers

One can “evaluate back” the pairing/unpairing tree representing the BDD, by using the pairing function `bitpair`. The inverse of `unPairingTree` is implemented as follows:

```
pairingTree(bdd(_,X),TT):-pairingTree1(X,TT).
```

```
pairingTree1(c(TT),TT).
pairingTree1(ite(_,L,R),TT):-
    pairingTree1(L,X),
    pairingTree1(R,Y),
    bitpair(X,Y,TT).
```

```
?- unPairingTree(3,42, BDD),pairingTree(BDD,N).
```

```
BDD = bdd(3,
    ite(2,
        ite(1,
            ite(0, c(0), c(0)),
            ite(0, c(0), c(0))),
        ite(1,
            ite(0, c(1), c(1)),
            ite(0, c(1), c(0))))),
N = 42
```


Note however that `pairingTree/2` does not act as an inverse of `reducedUnpairingTree/3`, given that the *structure* of the *BDD* tree is changed by reduction.

5.3 Boolean Evaluation of BDDs

This raises the obvious question: how can we recover the original truth table from a reduced BDD? The obvious answer is: by evaluating it as a boolean function! The predicate `ev/2` describes the *BDD* evaluator:

```
ev(bdd(NV,B),TT):-
    all_ones_mask(NV,M),
    eval_with_mask(NV,M,B,TT).

evc(0,_,0).
evc(1,M,M).

eval_with_mask(_,M,c(X),R):-evc(X,M,R).
eval_with_mask(NV,M,ite(X,T,E),R):-
    eval_with_mask(NV,M,T,A),
    eval_with_mask(NV,M,E,B),
    var_to_bitstring_int(NV,M,X,V),
    ite(V,A,B,R).
```

The predicate `ite/4` used in `eval_with_mask` implements the boolean function if *X* then *T* else *E* using arbitrary length bitvector operations:

```
ite(X,T,E, R):-R is xor(/\'(X,xor(T,E)),E).
```

Note that this equivalent formula for `ite` is slightly more efficient than the obvious one with \wedge and \vee as it requires only 3 boolean operations. We will use `ite/4` as the basic building block for implementing a boolean evaluator for BDDs.

5.4 The Equivalence

A surprising result is that boolean evaluation and structural transformation with repeated application of *pairing* produce the same result, i.e. the predicate `ev/2` also acts as an inverse of `reducedUnpairingTree/3` and `unPairingTree/2`.

As the following example shows, boolean evaluation ev/2 faithfully emulates pairingTree/2, on both plain and reduced BDDs.

```
?- unPairingTree(3,42,BDD),ev(BDD,N).
BDD = bdd(3,
    ite(2,
        ite(1,
            ite(0, c(0), c(0)),
            ite(0, c(0), c(0))),
        ite(1,
            ite(0, c(1), c(1)),
            ite(0, c(1), c(0))))),
```

N = 42

```
?- reducedUnpairingTree(3,42,BDD),ev(BDD,N).
BDD = bdd(3,
    ite(2,
        c(0),
        ite(1,
            c(1),
            ite(0, c(1), c(0))))),
N = 42
```

The main result of this subsection can now be summarized as follows:

Proposition 2 *Let B be the pairing/unpairing tree of depth N , obtained by recursive applications of `bitunpair` on a truth table T , as described by the predicate `unPairingTree(N,T,B)`.*

Then for any N and any T , when B is interpreted as an (unreduced) BDD, the result V of its boolean evaluation using the predicate `ev(N,B,V)` and the result R obtained by applying `pairingTree(N,B,R)` are both identical to T . Moreover, the operation `ev(N,B,V)` reverses the effects of both `unPairingTree` and `reducedUnpairingTree` with an identical result.

Proof sketch: The predicate `unPairingTree` builds a binary tree by splitting the bitstring $tt \in [0..2^N - 1]$ up to depth N . Observe that this corresponds to the Shannon expansion [4] of the formula associated to the truth table, using variable order $[n - 1, \dots, 0]$. Observe that the effect of `bitstring_unpair` is the same as

- the effect of `var_to_bitstring_int(N,M,(N-1),R)` acting as a mask selecting the left branch
- and the effect of its complement, acting as a mask selecting the right branch.

Given that 2^N is the double of 2^{N-1} , the same invariant holds at each step, as the bitstring length of the truth table reduces to half. On the other hand, it is clear that `ev` reverses the action of both `unPairingTree` and `reducedUnpairingTree` as BDDs and reduced BDDs represent the same boolean function [3].

This result can be seen as an intriguing isomorphism between boolean, arithmetic and symbolic computations.

6 Ranking and Unranking of BDDs

One more step is needed to extend the mapping between *BDDs* with N variables to a bijective mapping from/to *Nat*: we will have to “shift toward infinity” the starting point of each new block of BDDs in *Nat* as BDDs of larger and larger sizes are enumerated.

First, we need to know by how much - so we compute the sum of the counts of boolean functions with up to N variables.

```

bsum(0,0).
bsum(N,S):-N>0,N1 is N-1,bsum1(N1,S).

```

```

bsum1(0,2).
bsum1(N,S):-N>0,N1 is N-1,bsum1(N1,S1),S is S1+(1<<<(1<<N)).

```

The stream of all such sums can now be generated as usual:

```

bsum(S):-nat(N),bsum(N,S).

```

```

nat(0).
nat(N):-nat(N1),N is N1+1.

```

What we are really interested in, is decomposing N into the distance to the last bsum smaller than N, N_M and the index of that generates the sum, K.

```

to_bsum(N, X,N_M):-
    nat(X),bsum(X,S),S>N,!,
    K is X-1,
    bsum(K,M),
    N_M is N-M.

```

Unranking of an arbitrary BDD is now easy - the index K determines the number of variables and N_M determines the rank. Together they select the right BDD with `unPairingTree` and `reducedUnpairingTree/3`.

```

nat2unPairingTree(N,BDD):-to_bsum(N, K,N_M),unPairingTree(K,N_M,BDD).

```

```

nat2bdd(N,BDD):-to_bsum(N, K,N_M),reducedUnpairingTree(K,N_M,BDD).

```

The *Ranking* of a BDD is obtained as follows: we first compute its `NumberOfVars` and its rank `Nth`, then we shift the rank by the `bsums` up to `NumberOfVars`, enumerating the ranks previously assigned.

```

unPairingTree2nat(bdd(NumberOfVars,BDD),N) :-
    B=bdd(NumberOfVars,BDD),
    pairingTree(B,Nth),
    K is NumberOfVars-1,
    bsum(K,S),N is S+Nth.

```

```

bdd2nat(bdd(NumberOfVars,BDD),N) :-
    B=bdd(NumberOfVars,BDD),
    ev(B,Nth),
    K is NumberOfVars-1,
    bsum(K,S),N is S+Nth.

```

As the following example shows, `nat2unPairingTree/2` and `unPairingTree2nat/2` implement inverse functions.

```

?- nat2unPairingTree(42,BDD),unPairingTree2nat(BDD,N).
BDD = bdd(4,
    ite(3,
        ite(2,
            ite(1,

```

```

        ite(0, c(0), c(0)),
        ite(0, c(1), c(0))),
    ite(1,
        ite(0, c(1), c(0)),
        ite(0, c(0), c(0)))),
    ite(2,
        ite(1,
            ite(0, c(0), c(0)),
            ite(0, c(0), c(0))),
        ite(1, ite(0, c(0), c(0)),
            ite(0, c(0), c(0))))),

```

N = 42

The same applies to `nat2bdd/2` and its inverse `bdd2nat/2`.

```
?- nat2bdd(42,BDD),bdd2nat(BDD,N).
```

```

BDD = bdd(4,
    ite(3,
        ite(2,
            ite(1, c(0),
                ite(0, c(1), c(0))),
            ite(1,
                ite(0, c(1),c(0)), c(0))),
            c(0))),

```

N = 42

7 An application: BDD generators

BDD generators are useful for testing boolean circuits, boolean constraint solvers and learning algorithms involving BDDs.

7.1 Generating an infinite stream of BDDs

We can now generate infinite streams of BDDs as follows:

```
unPairingTree(BDD):-nat(N),nat2unPairingTree(N,BDD).
```

```
bdd(BDD):-nat(N),nat2bdd(N,BDD).
```

```

?- unPairingTree(BDD).
BDD = bdd(1, ite(0, c(0), c(0))) ;
BDD = bdd(1, ite(0, c(1), c(0))) ;
BDD = bdd(2, ite(1, ite(0, c(0), c(0)), ite(0, c(0), c(0)))) ;
BDD = bdd(2, ite(1, ite(0, c(1), c(0)), ite(0, c(0), c(0)))) ;
...
?- bdd(BDD).
BDD = bdd(1, c(0)) ;
BDD = bdd(1, ite(0, c(1), c(0))) ;
BDD = bdd(2, c(0)) ;

```

```

BDD = bdd(2, ite(1, ite(0, c(1), c(0)), c(0))) ;
BDD = bdd(2, ite(1, c(0), ite(0, c(1), c(0)))) ;
BDD = bdd(2, ite(0, c(1), c(0))) ;
...

```

7.2 Generating random BDDs

After defining:

```

ranbdd(NumberOfVars,BDD):-ranbdd(NumberOfVars,NumberOfVars,BDD).

```

```

ranbdd(MinVars,MaxVars,BDD):-
    MinVars1 is MinVars-1,
    bsum(MinVars1,Lower),
    bsum(MaxVars,Upper),
    Dif is Upper-Lower,
    ranint(Dif,Ran),
    N is Lower+Ran,
    nat2bdd(N,BDD).

```

```

ranint(Max,N):-random(X),N is floor(Max*X).

```

one can generate a uniform distribution of random BDDs with a given number of variables or a given range as follows:

```

?- repeat,ranbdd(1,2,BDD).
?- repeat,ranbdd(2,3,BDD).
BDD = bdd(2, ite(0, c(1), c(0))) ;
BDD = bdd(3, ite(2, ite(0, c(1), c(0)), c(0))) ;
BDD = bdd(3, ite(1, ite(0, c(1), c(0)), c(0))) ;
BDD = bdd(3, ite(2, ite(1, c(0), ite(0, c(1), c(0))), ite(0, c(1), c(0)))) .

```

8 Related work

A preliminary version of this paper is part of the CICLOPS'08 informal proceedings [9].

Pairing functions have been used in work on decision problems as early as [10, 11, 5]. *Ranking* functions can be traced back to Gödel numberings [1, 12] associated to formulae. Together with their inverse *unranking* functions they are also used in combinatorial generation algorithms [13, 2].

Binary Decision Diagrams are the dominant boolean function representation in the field of circuit design automation [14]. BDDs have been used in a Genetic Programming context [15, 16] as a representation of evolving individuals subject to crossovers and mutations expressed as structural transformations and recently in a machine learning context for compressing probabilistic Prolog programs [17] representing candidate theories.

Other interesting uses of BDDs in a logic and constraint programming context are related to representations of finite domains. In [18] an algorithm for finding minimal reasons for inferences is given.

9 Conclusion and Future Work

The surprising connection of pairing/unpairing functions and BDDs, is the indirect result of implementation work on a number of practical applications. Our initial interest has been triggered by applications of the encodings to combinational circuit synthesis in a logic programming framework [19, 20]. We have found them also interesting as uniform blocks for Genetic Programming applications of Logic Programming. In a Genetic Programming context [21], the bijections between bitvectors/natural numbers on one side, and trees/graphs representing BDDs on the other side, suggest exploring the mapping and its action on various transformations as a phenotype-genotype connection. Given the connection between BDDs to boolean and finite domain constraint solvers it would be interesting to explore in that context, efficient succinct data representations derived from our BDD encodings.

References

1. Gödel, K.: Über formal unentscheidbare Sätze der Principia Mathematica und verwandter Systeme I. Monatshefte für Mathematik und Physik **38** (1931) 173–198
2. Knuth, D.: The Art of Computer Programming, Volume 4, draft (2006) <http://www-cs-faculty.stanford.edu/~knuth/taocp.html>.
3. Bryant, R.E.: Graph-based algorithms for boolean function manipulation. IEEE Transactions on Computers **35**(8) (1986) 677–691
4. Shannon, C.E.: Claude Elwood Shannon: collected papers. IEEE Press, Piscataway, NJ, USA (1993)
5. Robinson, J.: General recursive functions. Proceedings of the American Mathematical Society **1**(6) (dec 1950) 703–718
6. Cégielski, P., Richard, D.: Decidability of the theory of the natural integers with the cantor pairing function and the successor. Theor. Comput. Sci. **257**(1-2) (2001) 51–77
7. Rosenberg, A.L.: Efficient pairing functions - and why you should care. International Journal of Foundations of Computer Science **14**(1) (2003) 3–17
8. Pigeon, S.: Contributions à la compression de données. Ph.d. thesis, Université de Montréal, Montréal (2001)
9. Tarau, P.: Pairing Functions, Boolean Evaluation and Binary Decision Diagrams. In Carro, M., Demoen, B., eds.: Proceedings of CICLOPS 2008, 8th International Colloquium on Implementation of Constraint and LOGic Programming Systems. (December 2008) 139–153
10. Peis, J.: Ein verfahren der mathematischen logik. The Journal of Symbolic Logic **3**(2) (jun 1938) 61–76
11. Kalmar, L.: On the reduction of the decision problem. first paper. ackermann prefix, a single binary predicate. The Journal of Symbolic Logic **4**(1) (mar 1939) 1–9
12. Hartmanis, J., Baker, T.P.: On simple goedel numberings and translations. In Loeckx, J., ed.: ICALP. Volume 14 of Lecture Notes in Computer Science., Springer (1974) 301–316

13. Martinez, C., Molinero, X.: Generic algorithms for the generation of combinatorial objects. In Rován, B., Vojtas, P., eds.: MFCS. Volume 2747 of Lecture Notes in Computer Science., Springer (2003) 572–581
14. Drechsler, R., Shi, J., Fey, G.: Synthesis of fully testable circuits from bdds. *IEEE Trans. on CAD of Integrated Circuits and Systems* **23**(3) (2004) 440–443
15. Sakanashi, H., Higuchi, T., Iba, H., Kakazu, Y.: Evolution of binary decision diagrams for digital circuit design using genetic programming. In Higuchi, T., Iwata, M., Liu, W., eds.: ICES. Volume 1259 of Lecture Notes in Computer Science., Springer (1996) 470–481
16. Chen, S.T., Lin, S.S., Huang, L.T., Wei, C.J.: Towards the exact minimization of bdds-an elitism-based distributed evolutionary algorithm. *J. Heuristics* **10**(3) (2004) 337–355
17. Raedt, L.D., Kersting, K., Kimmig, A., Revoredo, K., Toivonen, H.: Compressing probabilistic prolog programs. *Machine Learning* **70**(2-3) (2008) 151–168
18. Hawkins, P., Stuckey, P.J.: A hybrid bdd and sat finite domain constraint solver. In Hentenryck, P.V., ed.: PADL. Volume 3819 of Lecture Notes in Computer Science., Springer (2006) 103–117
19. Tarau, P., Luderman, B.: Exact combinational logic synthesis and non-standard circuit design. In: CF '08: Proceedings of the 2008 conference on Computing frontiers, New York, NY, USA, ACM (2008) 179–188
20. Tarau, P., Luderman, B.: A Logic Programming Framework for Combinational Circuit Synthesis. In: 23rd International Conference on Logic Programming (ICLP), LNCS 4670, Porto, Portugal, Springer (September 2007) 180–194
21. Koza, J.R.: Genetic Programming: On the Programming of Computers by Means of Natural Selection. MIT Press, Cambridge, MA, USA (1992)

Appendix

To make the code in the paper fully self contained, we list here some auxiliary functions.

```
% converts an int to a list of bits, least significant first
to_rbits(0, []).
to_rbits(N, [B|Bs]):-N>0, B is N mod 2, N1 is N//2,
    to_rbits(N1, Bs).

% converts a list of bits (least significant first) into an int
from_rbits(Rs, N):-nonvar(Rs), from_rbits(Rs, 0, 0, N).

from_rbits([], _, N, N).
from_rbits([X|Xs], E, N1, N3):-NewE is E+1, N2 is X<<E+N1,
    from_rbits(Xs, NewE, N2, N3).
```

The following figures visualize properties of our pairing/unpairing functions. Given that unpairing functions are bijections from Nat to $Nat \times Nat$ they will




Fig. 1: Graph obtained by recursive application of `bitunpair` for 2008

progressively cover all points having natural number coordinates in their range in the plane. Figure 2 show the curve generated by `bitunpair`. Fig. 3 shows the action of the pairing function `bitpair` on its *two* arguments x, y in $[0..63]$.




Fig. 2: 2D curve connecting values of `bitunpair n` for $n \in [0..2^{10} - 1]$




Fig. 3: Values of `bitpair x y` with x, y in $[0..63]$