

Synthesis of Modality Definitions and a Theorem Prover for Epistemic Intuitionistic Logic

Paul Tarau

Department of Computer Science and Engineering
University of North Texas
paul.tarau@unt.edu

Abstract. We propose a mechanism for automating discovery of definitions, that, when added to a logic system for which we have a theorem prover, extends it to support an embedding of a new logic system into it. As a result, the synthesized definitions, when added to the prover, implement a prover for the new logic.

As an instance of the proposed mechanism, we derive a Prolog theorem prover for an interesting but unconventional epistemic Logic by starting from the sequent calculus **G4IP** that we extend with operator definitions to obtain an embedding in intuitionistic propositional logic (**IPC**). With help of a candidate definition formula generator, we discover epistemic operators for which axioms and theorems of Artemov and Protopopescu’s *Intuitionistic Epistemic Logic* (**IEL**) hold and formulas expected to be non-theorems fail.

We compare the embedding of **IEL** in **IPC** with a similarly discovered successful embedding of Dosen’s double negation modality, judged inadequate as an epistemic operator. Finally, we discuss the failure of the *necessitation rule* for an otherwise successful **S4** embedding and share our thoughts about the intuitions explaining these differences between epistemic and alethic modalities in the context of the Brouwer-Heyting-Kolmogorov semantics of intuitionistic reasoning and knowledge acquisition.

Keywords: automatic synthesis of logic systems, deriving new theorem provers via program synthesis, epistemic intuitionistic logic, propositional intuitionistic logic, Prolog-based theorem provers, embedding of modal logics into intuitionistic logic.

1 Introduction

Deriving new logic systems and discovering relationships between them not only requires a knowledge-intensive understanding of the intricate connections between their axioms and inference rules but it is also a time-intensive trial and error process for the human logician. This is especially the case for logic systems that depart from the usual expectations coming from the prevalent use of classical logic in today’s computational tools and methodologies as well from familiarity with more commonly used forms of modal logic (e.g., alethic, temporal).

This motivates our effort to explore ways to automate this process, resulting not only in discovering some salient relationships between new and well-established logic

38 systems, but also in software artifacts (e.g., automated theorem provers) facilitating
39 reasoning in these less explored new logics.

40 Epistemic Logic Systems have been derived often in parallel and sometime as af-
41 terthoughts of alethic Modal Logic Systems, in which modalities are defined by axioms
42 and additional inference rules extending classical logic.

43 In the context of Answer Set Programming (**ASP**) epistemic logics hosted in this
44 framework like e.g., [1–3] show that *intermediate logics*¹ (e.g., equilibrium logic, [4])
45 can express epistemic operators by extending them with definitions of epistemic opera-
46 tors.

47 Steps², further below classical logic or **ASP**, are taken in recent work [5], based
48 on the Brouwer-Heyting-Kolmogorov (**BHK**) view of intuitionistic logic that takes into
49 account the constructive nature of knowledge, modeling more accurately the connection
50 between proof systems and the related mental processes. Along these lines, our inquiry
51 into epistemic logic will focus on knowledge vs. truth seen as intuitionistic provability.

52 Like in the case of embedding epistemic operators into **ASP** systems, but with au-
53 tomation in mind, we will design a synthesis mechanism for epistemic operators via
54 embedding in **IPC**. For this purpose we will generate *candidate formulas* that verify
55 axioms, theorems and rules and fail on expected non-theorems. For this purpose, we
56 will use a lightweight **IPC theorem prover** and we will also show that this view gener-
57 alizes to a mechanism for discovering for a given modal logic, when possible, a simple
58 embedding of the logic into **IPC** and derivation of a theorem prover for it.

59 Our starting point is Artemov and Protopopescu’s *Intuitionistic Epistemic Logic*
60 (**IEL**) [5] that will provide the axioms, theorems and non-theorems stating the require-
61 ments that must hold for the definitions extending **IPC**. The discovery mechanism will
62 also bring up Dosen’s interpretation of double negation [6] as a potential epistemic
63 operator and we will look into applying the same discovery mechanisms to find an em-
64 bedding of modal logic **S4** in **IPC**, with special focus on the impact of the *necessitation*
65 *rule*, which requires that all theorems of the logic are necessarily true.

66 **The rest of the paper is organized as follows.** Section 2 overviews Artemov and
67 Protopopescu’s *Intuitionistic Epistemic Logic (IEL)*. Section 3 introduces the **G4IP** se-
68 quent calculus prover for Intuitionistic Propositional Logic (**IPC**). Section 4 describes
69 the generator for candidate formulas extending **IPC** with modal operator definitions.
70 Section 5 explains the discovering of the definitions that ensure the embedding of **IEL**
71 into **IPC** and the discovering of the embedding of Dosen’s double negation as a modal-
72 ity operator. It also discusses the intuitions behind the embedding of **IEL**, including the
73 epistemic equivalent of the necessity rule, in **IPC** and the adequacy of this embedding
74 as a constructive mechanism for reasoning about knowledge. Section 6 studies the case
75 of the **S4** modal logic and the failure of the necessity rule, indicating the difficulty of
76 embedding it in **IPC** by contrast to **IEL**. Section 7 overviews some related work and
77 section 8 concludes the paper.

¹ Logics stronger than intuitionistic but weaker than classical.

² Actually infinitely many, as there’s an infinite lattice of intermediate logics between classical and intuitionistic logic.

78 The paper is written as a literate SWI-Prolog program with its extracted code at
79 <https://raw.githubusercontent.com/ptarau/TypesAndProofs/master/ieltp.pro>.

80 2 Overview of Artemov and Protopopescu’s IEL logic

81 In [5] a system for Intuitionistic Epistemic Logic is introduced that

82 “maintains the original Brouwer-Heyting-Kolmogorov semantics for intuition-
83 ism and is consistent with the well-known approach that intuitionistic knowl-
84 edge be regarded as the result of verification”.

85 Instead of the classic, alethic-modalities inspired **K** operator for which

$$\mathbf{KA} \rightarrow A$$

86 Artemov and Protopopescu argue that *co-reflection* expresses better the idea of *con-*
87 *structivity of truth*

$$A \rightarrow \mathbf{KA}$$

88 They also argue that this applies to both belief and knowledge i.e., that

89 “The verification-based approach allows that justifications more general than
90 proof can be adequate for belief and knowledge”.

91 On the other hand, they consider *intuitionistic reflection* acceptable, expressing the
92 fact that “known propositions cannot be false”:

$$\mathbf{KA} \rightarrow \neg\neg A$$

93 Thus, they position intuitionistic knowledge of A between A and $\neg\neg A$ and given
94 that (via Glivenko’s transformation [7]) applying double negation to a formula embeds
95 classical propositional calculus into **IPC**, they express this view as:

$$96 \quad \textit{Intuitionistic Truth} \Rightarrow \textit{Intuitionistic Knowledge} \Rightarrow \textit{Classical Truth}.$$

97 They axiomatize the system **IEL** as follows.

- 98
- 99 1. Axioms of propositional intuitionistic logic;
 - 100 2. $\mathbf{K}(A \rightarrow B) \rightarrow (\mathbf{KA} \rightarrow \mathbf{KB})$; (distribution)
 - 101 3. $A \rightarrow \mathbf{KA}$. (co-reflection)
 - 102 4. $\mathbf{KA} \rightarrow \neg\neg A$ (intuitionistic reflection)

103

104 **Rule Modus Ponens.**

105 They also argue that a weaker logic of belief (**IEL**[−]) is expressed by considering
106 only axioms **1,2,3**.

107 3 The G4ip prover for IPC

108 We will describe next our lightweight propositional intuitionistic theorem prover, that
109 will be used to discover an embedding of **IEL** into **IPC**.

110 3.1 The LJT/G4ip calculus, (restricted here to the implicational fragment)

111 Motivated by problems related to loop avoidance in implementing Gentzen's **LJ** calcu-
 112 lus, Roy Dyckhoff [8] introduces the following rules for the **G4ip** calculus³.

$$113 \quad LJT_1 : \quad \overline{A, \Gamma \vdash A}$$

$$114 \quad LJT_2 : \quad \frac{A, \Gamma \vdash B}{\Gamma \vdash A \rightarrow B}$$

$$115 \quad LJT_3 : \quad \frac{B, A, \Gamma \vdash G}{A \rightarrow B, A, \Gamma \vdash G}$$

$$116 \quad LJT_4 : \quad \frac{D \rightarrow B, \Gamma \vdash C \rightarrow D \quad B, \Gamma \vdash G}{(C \rightarrow D) \rightarrow B, \Gamma \vdash G}$$

117 Note that LJT_4 ensures termination as formulas in the sequent become smaller in a
 118 multiset ordering. The rules work with the context Γ being either a multiset or a set,
 119 and the calculus is sound and complete for IPC.

120 For supporting negation, one also needs to add LJT_5 that deals with the special term
 121 *false*. Then negation of A is defined as $A \rightarrow \text{false}$.

$$122 \quad LJT_5 : \quad \overline{\text{false}, \Gamma \vdash G}$$

123 Rules for conjunction, disjunction and bi-conditional (not shown here) are also part of
 124 the calculus.

125 As it is not unusual with logic formalisms, the same calculus had been discovered
 126 independently in the 1950's by Vorob'ev and in the 80's-90's by Hudelmaier [9, 10].

140 3.2 A Lightweight Theorem Prover for Intuitionistic Propositional Logic

141 Starting from the sequent calculus for the intuitionistic propositional logic in G4ip [8],
 142 to which we have also added rules for the " \leftrightarrow " relation, we obtain the following
 143 lightweight **IPC** prover.

```
144 :- op(525, fy, ~ ).
145 :- op(550, xfy, & ).    % right associative
146 :- op(575, xfy, v ).    % right associative
147 :- op(600, xfx, <-> ). % non associative
```

```
148 prove_in_ipc(T) :- prove_in_ipc(T, []).
```

³ Originally called the LJT calculus in [8]. Restricted here to its key implicational fragment.

```

150 prove_in_ipc(A,Vs):-memberchk(A,Vs),!.
151 prove_in_ipc(_,Vs):-memberchk(false,Vs),!.
152 prove_in_ipc(A<->B,Vs):-!,prove_in_ipc(B,[A|Vs]),prove_in_ipc(A,[B|Vs]).
153 prove_in_ipc((A->B),Vs):-!,prove_in_ipc(B,[A|Vs]).
154 prove_in_ipc(A & B,Vs):-!,prove_in_ipc(A,Vs),prove_in_ipc(B,Vs).
155 prove_in_ipc(G,Vs1):- % atomic or disj or false
156     select(Red,Vs1,Vs2),
157     prove_in_ipc_reduce(Red,G,Vs2,Vs3),
158     !,
159     prove_in_ipc(G,Vs3).
160 prove_in_ipc(A v B, Vs):- (prove_in_ipc(A,Vs);prove_in_ipc(B,Vs)),!.

```

```

161 prove_in_ipc_reduce((A->B),_,Vs1,Vs2):-!,prove_in_ipc_imp(A,B,Vs1,Vs2).
162 prove_in_ipc_reduce((A & B),_,Vs,[A,B|Vs]):-!.
163 prove_in_ipc_reduce((A<->B),_,Vs,[(A->B),(B->A)|Vs]):-!.
164 prove_in_ipc_reduce((A v B),G,Vs,[B|Vs]):-(prove_in_ipc(G,[A|Vs])).

```

```

165 prove_in_ipc_imp((C->D),B,Vs,[B|Vs]):-!,prove_in_ipc((C->D),[(D->B)|Vs]).
166 prove_in_ipc_imp((C & D),B,Vs,[(C->(D->B))|Vs]):-!.
167 prove_in_ipc_imp((C v D),B,Vs,[(C->B),(D->B)|Vs]):-!.
168 prove_in_ipc_imp((C<->D),B,Vs,[(C->D)->((D->C)->B)|Vs]):-!.
169 prove_in_ipc_imp(A,B,Vs,[B|Vs]):-(memberchk(A,Vs)).

```

170 We validate it first by testing it on the implicational subset, derived via the Curry-
171 Howard isomorphism [11], then against Roy Dyckhoff's Prolog implementation⁴, work-
172 ing on formulas up to size 12. Finally we run it on human-made tests⁵, on which we get
173 no errors, solving correctly 161 problems, with a 60 seconds timeout, compared with
174 the 175 problems solved by Roy Dyckhoff's more refined, heuristics-based 400 lines
175 prover, with the same timeout⁶. We refer to [11] for the derivation steps of variants
176 of this prover working on the implicational and nested Horn clause fragments of **IPC**.
177 While more sophisticated tableau-based provers are available for **IPC** among which we
178 mention the excellent Prolog-based fCube [12], our prover's compact size and adequate
179 performance will suffice ⁷.

180 4 The definition formula generator

181 We start with a candidate formula generator that we will constrain further to be used for
182 generating candidate definitions of our modal operators.

⁴ https://github.com/ptarau/TypesAndProofs/blob/master/third_party/dyckhoff_orig.pro

⁵ at <http://iltp.de>

⁶ <https://github.com/ptarau/TypesAndProofs/blob/master/tester.pro>

⁷ In fact, our prover is faster than both fCube and Dyckhoff's prover on the set of formulas of small size on which our definition induction algorithm will run.

183 4.1 Generating Operator Trees

184 We generate all formulas of a given size by decreasing the available size parameter at
 185 each step when nodes are added to a tree representation of a formula. Prolog's **DCG**
 186 mechanism is used to collect the leaves of the tree.

```
187 genOperatorTree(N,Ops,Tree,Leaves):-
188     genOperatorTree(Ops,Tree,N,0,Leaves,[]).
189
190 genOperatorTree(_,V,N,N)-->[V].
191 genOperatorTree(Ops,OpAB,SN1,N3)-->
192     { SN1>0,N1 is SN1-1,
193       member(Op,Ops),make_oper2(Op,A,B,OpAB)
194     },
195     genOperatorTree(Ops,A,N1,N2),
196     genOperatorTree(Ops,B,N2,N3).
197
198 make_oper2(Op,A,B,OpAB):-functor(OpAB,Op,2),arg(1,OpAB,A),arg(2,OpAB,B).
```

199 4.2 Synthesizing the definitions of modal operators

200 As we design a generic definition discovery mechanism, we will denote generically our
 201 modal operators as follows.

- 202 - “#” for “□”=necessary and “K”=known
- 203 - “*” for “◇”=possible and “M”=knowable

204 After the operator definitions

```
205 :- op( 500, fy, #).
206 :- op( 500, fy, *).
```

207 we specify our generator as covering the usual binary operators and we constrain it to
 208 have at least one of the leaves of its generated trees to be a variable. Besides the **false**
 209 constant used in the definition of negation, we introduce also a new constant symbol “?”
 210 assumed not to occur in the language. Its role will be left unspecified until the possible
 211 synthesized definitions will be filtered. We will constrain candidate definitions to ensure
 212 that axioms and selected theorems hold and selected non-theorems fail.

```
213 genDef(M,Def):-genDef(M,[(->),( & ),(v)], [false,?],Def).
214
215 genDef(M,Ops,Cs, (#(X):-T)):-
216     between(0,M,N),
217     genOperatorTree(N,Ops,T,Vs),
218     pick_leaves(Vs,[X|Cs]),
219     term_variables(Vs,[X]).
```

220 Leaves of the generated trees will be picked from a given set.

```
221 pick_leaves([],_).
222 pick_leaves([V|Vs],Ls):-member(V,Ls),pick_leaves(Vs,Ls).
```

223 We first expand our operator definitions for the “ \sim ” negation and “ $*$ ” modal operator
 224 while keeping atomic variables and the special constant `false` untouched.

```
225 expand_defs(_,false,R) :-!,R=false.
226 expand_defs(_,A,R) :-atomic(A),!,R= A.
227 expand_defs(D,~(A),(B->false)) :-!,expand_defs(D,A,B).
228 expand_defs(D,*A,R):-!,expand_defs(D,~ (# (~A)),R).
```

229 The special case for expanding a candidate operator definition `D` requires a fresh variable
 230 for each instance, ensured by Prolog’s built-in `copy_term`.

```
231 expand_defs(D,#(X),R) :-!,copy_term(D,(#(X):-T)),expand_defs(D,T,R).
```

232 Other operators are traversed generically by using Prolog’s “`=..`” built-in and by re-
 233 cursing with `expand_def_list` on their arguments.

```
234 expand_defs(D,A,B) :-
235     A=..[F|Xs],
236     expand_def_list(D,Xs,Ys),
237     B=..[F|Ys].
```

```
238 expand_def_list(_,[],[]).
239 expand_def_list(D,[X|Xs],[Y|Ys]) :-
240     expand_defs(D,X,Y),
241     expand_def_list(D,Xs,Ys).
```

242 The predicate `prove_with_def` refines our **G4ip** prover by first expanding the defini-
 243 tions extending **IPC** with a given candidate modality.

```
244 prove_with_def(Def,T0) :-expand_defs(Def,T0,T1),prove_in_ipc(T1,[]).
```

245 The definition synthesizer will filter the candidate definitions provided by `genDef` such
 246 that the predicate `prove_with_def` succeeds on all theorems and fails on all non-
 247 theorems, provided as names of the facts of arity 1 containing them.

```
248 def_synth(M,D):-def_synth(M,iel_th,iel_nth,D).
249
250 def_synth(M,Th,NTh,D):-
251     genDef(M,D),
252     forall(call(Th,T),prove_with_def(D,T)),
253     forall(call(NTh,NT),\+prove_with_def(D,NT)).
```

254 Note that the generator first builds smaller formulas and then larger ones up the specified
 255 maximum size.

256 **Example 1** *Candidate definitions up to size 2*

```
257 ?- forall(genDef(2,Def),println(Def)).
258 #A :- A
259 #A :- A -> A
260 #A :- A -> false
261 #A :- A -> ?
262 #A :- false -> A
263 #A :- ? -> A
```

```

264 #A :- A & A
265 #A :- A & false
266 #A :- A & ?
267 ...
268 #A :- (A -> ?) -> A
269 ...
270 #A :- (? v A) v ?
271 #A :- (? v false) v A
272 #A :- (? v ?) v A

```

273 5 Discovering the embedding of IEL and Dosen's double negation 274 modality in IPC

275 We specify a given logic (e.g., **IEL** or **S4**) by stating theorems on which the prover
276 extended with the synthetic definition should succeed and non-theorems on which it
277 should fail.

278 5.1 The discovery mechanism for IEL

279 We start with the axioms of Artemov and Protopopescu's **IEL** system:

```

280 iel_th(a -> # a).
281 iel_th(# (a->b)->(# a-> # b)).
282 iel_th(# a -> ~ ~ a).

```

283 Note that the axioms would be enough to specify the logic, but we also add some the-
284 orems when intuitively relevant and/or mentioned in [5]. Our Prolog code, running in
285 less than a second, is not slowed down by this in any significant way.

```

286 iel_th(# (a & b) <-> (# a & # b)).
287 iel_th(~ # false).
288 iel_th(~ (# a & ~ a)).
289 iel_th(~a -> ~ # a).
290 iel_th( ~ ~ (# a -> a)).
291 iel_th(# a & # (a->b) -> # b).
292 iel_th(* (a & b) <-> (* a & * b)).
293 iel_th(# a -> * a).
294 iel_th(# a v # b -> # (a v b) ).
295 iel_th(# p <-> # # p).
296 iel_th(* a <-> * * a).
297 iel_th(a -> *a).

```

298 Again, following [5], we add our non-theorems.

```

299 iel_nth(# a -> a).
300 iel_nth(# (a v b) -> # a v # b).
301 iel_nth(# a).
302 iel_nth(~ (# a)).
303 iel_nth(# false).
304 iel_nth(# a).

```



```

305 iel_nth(~ (# a)).
306 iel_nth(* false).

```

307 The *necessitation rule* in a modal logic requires that if T is a theorem than $\#T$ is also a
 308 theorem. This expresses the fact that the theorems of the logic are *necessarily* true, or
 309 in an epistemic context that if T is an (intuitionistically proven) theorem, then the agent
 310 *knows* T . Thus, we define (implicit) facts via a Prolog rule that states that the (generic)
 311 necessity operator “ $\#$ ” applied to proven theorems or axioms generates new theorems.

```

312 iel_nec_th(T):-iel_th(T).
313 iel_nec_th(# T):-iel_th(T).

```

314 Finally, we obtain the discovery algorithm for **IEL** formula definitions and for **IEL**
 315 extended with the necessitation rule.

```

316 iel_discover:-
317     backtrack_over((def_synth(2,iel_th,iel_nth,D),println(D))).
318
319 iel_nec_discover:-
320     backtrack_over((def_synth(2,iel_nec_th,iel_nth,D),println(D))).
321
322 backtrack_over(Goal):-call(Goal),fail;true.
323
324 println(T):-numbervars(T,0,_),writeln(T).

```

325 We run `iel_discover`, ready to see the surviving definition candidates.

326 **Example 2** *Definition discovery without the necessitation rule.*

```

327 ?- iel_discover.
328 #A:-(A->false)->A
329 #A:-(A->false)->false
330 #A:-(A-> ?)->A
331 true.

```

332 **Example 3** *Definition discovery with the necessitation rule.*

```

333 ?- iel_nec_discover.
334 #A:-(A->false)->A
335 #A:-(A->false)->false
336 #A:-(A-> ?)->A
337 true.

```

338 Unsurprisingly, the results are the same, as a consequence of axiom $A \rightarrow \#A$.

339 Clearly, the formula $\#A:-(A \rightarrow \text{false}) \rightarrow A$ is not interesting as it would define
 340 knowing something as a contradiction that implies itself.

341 This brings us to the second definition formula candidate.

342 **5.2 Eliminating Dosen’s double negation modality**

343 In [2] double negation in IPC is interpreted as a “ \Box ” modality. This corresponds to one
 344 of the synthetic definitions $\#A :- (A \rightarrow \text{false}) \rightarrow \text{false}$ that is equivalent in **IPC** to

345 $\#A :- \sim\sim A$. It is argued in [5] that it does not make sense as an epistemic modality,
 346 mostly because it would entail that all classical theorems are known intuitionistically.

347 We eliminate it by requiring the collapsing of “*” into “#” to be a non-theorem:

348 `iel_nth(* a <-> # a).`

349 In fact, while *known* (#) implies *knowable* ($\sim\sim = *$), it is reasonable to think, as in
 350 most modal logics, that the inverse implication does not hold.

351 After that, we have:

352 **Example 4** *The double negation modality is eliminated, as it collapses # and *.*

353 `?- iel_discover.`

354 `#A:-(A -> ?)->A`

355 `true.`

356

357 `?- iel_nec_discover.`

358 `#A:-(A -> ?)->A`

359 `true.`

360 5.3 Knowledge as awareness?

361 This leaves us with the $\#A :- (A \rightarrow ?) \rightarrow A$.

362 Among the consequences of the fact that intuitionistic provability strictly implies
 363 classical, is that there’s plenty of room left between p and $\sim\sim p$, where both # and * find
 364 their place, given that the following implication chain holds.

365 $p \rightarrow \#p \rightarrow *p \rightarrow \sim\sim p$

366 Let us now find an (arguably) intuitive meaning for the “?” constant in the definition.
 367 The interpretation of knowledge as awareness about truth goes back to [13]. Our final
 368 definition of intuitionistic epistemic modality as “ $\#A :- (A \rightarrow ?) \rightarrow A$ ” suggests
 369 interpreting “?” as awareness of an agent entailed by (a proof of) A . With this in mind,
 370 one obtains an embedding of **IEL** in **IPC** via the extension

$$\mathbf{KA} \equiv (A \rightarrow \mathbf{eureka}) \rightarrow A$$

371 where **eureka** is a new symbol not occurring in the language⁸.

372 In line with the Brouwer-Heyting-Kolmogorov (**BHK**) interpretation of intuitionis-
 373 tic proof, we may say that an agent *knows* A if and only if A is validated by a proof of A
 374 that induces awareness of the agent about it.

375 Thus knowledge of an agent, in this sense, collects facts that are proven construc-
 376 tively in a way that is “understood” by the agent. The consequence

$$\mathbf{KA} \rightarrow \neg\neg A$$

377 would then simply say that intuitionistic truths, that the agent is aware of, are also
 378 classically valid.

379 Thus, we can define our *newly synthesized* prover for **IEL** as follows.

⁸ Not totally accidentally named, given the way Archimedes expressed his sudden *awareness*
 about the volume of water displaced by his immersed body.

```
380 iel_prove(P):-prove_with_def((#A :- (A -> eureka) -> A),P).
```

381 Interestingly, if one allows **eureka** to occur in the formulas of the language given as
 382 input to the prover, then it becomes (the unique) value for which we have equivalence
 383 between being known and having a proof.

```
384 ?- iel_prove(#eureka <-> eureka).
385 true .
```

386 Similarly, it would also follow that

```
387 ?- iel_prove(*eureka <-> ~ ~ eureka).
388 true.
```

389 Thus, one would need to forbid accepting it as part of the prover’s language to closely
 390 follow the intended semantics of **IEL**.

391 5.4 Discussion

392 *The most significant consequence of the successful embedding of **IEL** into **IPC** via the*
 393 *epistemic modality definition #A :- (A -> eureka) -> A) is that we have actually*
 394 *derived a theorem prover for **IEL**.* The theorem prover is implemented by the predicate
 395 `iel_prove/1` by extending a theorem prover for **IPC** with the induced definition.

396 As the **IPC** fragment with two variables, implication and negation has exactly **518**
 397 equivalence classes of formulas [14, 15], one would expect the construction deriving
 398 “*” from “#” to reach a fixpoint. We can use our prover to find out when that happens.

```
399 ?- iel_prove(#p <-> ~ # (~p)).
400 false.
401 iel_prove(*p <-> ~(~p)).
402 true.
```

403 Thus the fixpoint of the construction is “*”, that we have interpreted as meaning that a
 404 proposition is *knowable*. Therefore, the equivalence reads reasonably as “something is
 405 knowable if and only if its negation is not knowable”. Note also that

```
406 ?- iel_prove(~(*(~p)) -> #p).
407 false.
```

408 fails, by contrast to the equivalence $\Box p \equiv \neg \Diamond \neg p$ usual in classical modal logics.

409 6 Discovering an embedding of **S4** without the necessitation rule

410 The fact that both **IPC** and **S4** are known to be PSPACE-complete [16] means that
 411 polynomial-time translations exist between them.

412 In fact, Gödel’s translation from **IPC** to **S4** (by prefixing each subformula with the
 413 \Box operator) shows that the embedding of **IPC** into **S4** can be achieved quite easily, by
 414 using purely syntactic means. However, the (very) few papers attempting the inverse
 415 translation [17, 18] rely on methods often involving intricate semantic constructions.

416 We will use our definition generator to identify the problem that precludes a simple
 417 embedding of **S4** into **IPC**.

418 We start with the axioms of **S4**.

```

419 s4_th(# a -> a).
420 s4_th(# (a->b) -> (# a -> # b)).
421 s4_th(# a -> # # a).

```

422 We add a few theorems.

```

423 s4_th(* * a <-> * a).
424 s4_th(a -> * a).
425 s4_th(# a -> * a).
426 s4_th(# a v # b -> # (a v b)).
427 s4_th(# (a v b) -> # a v # b).

```

428 We add some non-theorems that ensure additional filtering.

```

429 s4_nth(# a).
430 s4_nth(~ (# a)).
431 s4_nth(# false).
432 s4_nth(* false).
433 s4_nth(* a -> # * a). % true only in S5
434 s4_nth(a -> # a).
435 s4_nth(* a -> a).
436 s4_nth(# a <-> ?).
437 s4_nth(* a <-> ?).

```

438 Like in the case of **IEL** we define implicit facts stating that the necessitation rule holds.

```

439 s4_nec_th(T):-s4_th(T).
440 s4_nec_th(# T):-s4_th(T).

```

441 Finally we implement the definition discovery predicates and run them.

```

442 s4_discover:-
443   backtrack_over((def_synth(2,s4_th,s4_nth,D),println(D))).
444
445 s4_nec_discover:-
446   backtrack_over((def_synth(2,s4_nec_th,s4_nth,D),println(D))).

```

447 **Example 5** *The necessitation rule eliminates all simple embeddings of **S4** into **IPC**, while a lot of definition formulas pass without it.*

```

449 ?- s4_discover.
450 #A :- A & ?
451 #A :- ? & A
452 #A :- A & (A-> ?)
453 #A :- A & (? -> false)
454 ...
455 true.
456
457 ?- s4_nec_discover.
458 true.

```

459 Among the definitions succeeding without passing the necessity rule test, one might want to pick `#A :- ? & A` as an approximation of the **S4** “ \Box ” operator. In this case “?”

would simply state that “the IPC prover is sound and complete”. Still, given the failure of the necessitation rule, the resulting logic is missing a key aspect of the intended meaning of **S4**-provability.

7 Related work

Program synthesis techniques have been around in logic programming with the advent of Inductive Logic Programming [19], but the idea of learning Prolog programs from positive and negative examples goes back to [20]. Our definition synthesizer fits in this paradigm, with focus on the use of a theorem prover of a decidable logic (**IPC**) filtering formulas provided by a definition generator through theorems as positive examples and non-theorems as negative examples. The means we use for our definition synthesis are in fact as simple as those described in [20]. The strength of our approach comes from the use of a theorem prover that efficiently validates or rejects definition candidates. The idea to use the new constant “?” in our synthesizer is inspired by proofs that some fragments of **IPC** reduced to two variables have a (small) finite number of equivalence classes [14, 15] as well as by the introduction of new variables, in work on polynomial embeddings of **S4** into **IPC** [17, 18].

We refer to [5] for a thorough discussion of the merits of **IEL** compared to epistemic logics following closely classical modal logic, but the central idea about using intuitionistic logic is that of *belief and knowledge as the product of verification*. Our embedding of **IEL** in **IPC** can be seen as a simplified view of this process through a generic “awareness of an agent” concept in line with [13].

In [1] the concept of *epistemic specifications* is introduced that support expressing knowledge and belief in an Answer Set Programming framework. Interestingly, refinements of this work like [21] and [3] discuss difficulties related to expressing an assumption like $p \rightarrow \mathbf{K}p$ in terms of **ASP-based** epistemic operators.

Equilibrium logic [4] gives a semantics to Answer Set programs by extending the 3-valued intermediate logic of here-and-there **HT** with Nelson’s constructive strong negation. In [22] a 5-valued truth-table semantics for equilibrium logic is given. In [23] (and several other papers) epistemic extensions of equilibrium logic [4] are proposed, in which $\mathbf{K}p \rightarrow p$. By contrast to “alethic inspired” epistemic logics postulating $\mathbf{K}p \rightarrow p$ we closely follow the $p \rightarrow \mathbf{K}p$ view on which [5] is centered.

While we have eliminated Dosen’s double negation modality [6] as an epistemic operator $\mathbf{K}p \equiv \neg\neg p$, it is significant that it came out as the only other meaningful candidate produced by our definition synthesizer.

This suggests that it might be worth investigating further how a similar definition discovery mechanism as the one we have used for **IEL** and **S4** would work for logics with multiple negation operators like equilibrium logic.

Besides the $\mathbf{K}p \rightarrow p$ vs. $p \rightarrow \mathbf{K}p$ problem a more general question is the choice of the logic supporting the epistemic operators, among logics with finite truth-value models (e.g., classical logic or equilibrium logic) or, at the limit, intuitionistic logic itself, with no such models. Arguably, this could be application dependent, as epistemic operators built on top of IPC are likely to fit better the landscape with intricate nuances of a richer set of epistemic and doxastic operators, while such operators built on top of

504 finite-valued intermediate logics would benefit from simpler decision procedures and
505 faster evaluation mechanisms.

506 8 Conclusions

507 We have devised a general mechanism for synthesizing definitions that extend a given
508 logic system endowed with a theorem prover. The set of theorems on which the ex-
509 tended prover should succeed and the set of non-theorems on which it should fail can
510 be seen as a declarative specification of the extended system. Success of the approach
511 on embedding the **IEL** system in **IPC** and failure on trying to embed **S4** has revealed
512 the individual role of the axioms, theorems and rules that specify a given logic system
513 and their interaction with the necessitation rule .

514 Given its generality, our definition generation technique can be applied also to epis-
515 temic or modal logic axiom systems to find out if they have interesting embeddings in
516 **ASP** and superintuitionistic logics for which high quality solvers or theorem provers
517 exist. Our program synthesis process, when the embedding succeeds, provides a way to
518 automate the exploration of a new logic system with help of its derived theorem prover
519 and facilitates the work of the human logician to validate or invalidate the intuitions
520 behind it.

521 Acknowledgement

522 We thank the participants to the **EELP'2019** workshop⁹ for their constructive sugges-
523 tions and comments.

524 References

- 525 1. Gelfond, M.: Strong Introspection. In: Proceedings of the Ninth National Conference on
526 Artificial Intelligence - Volume 1. AAAI'91, AAAI Press (1991) 386–391
- 527 2. Baral, C., Gelfond, G., Son, T.C., Pontelli, E.: Using answer set programming to model
528 multi-agent scenarios involving agents' knowledge about other's knowledge. In: Proceedings
529 of the 9th International Conference on Autonomous Agents and Multiagent Systems: Volume
530 1 - Volume 1. AAMAS '10, Richland, SC, International Foundation for Autonomous Agents
531 and Multiagent Systems (2010) 259–266
- 532 3. Shen, Y.D., Eiter, T.: Evaluating epistemic negation in answer set programming. *Artif. Intell.*
533 **237(C)** (August 2016) 115–135
- 534 4. Pearce, D.: A new logical characterisation of stable models and answer sets. In: Selected
535 Papers from the Non-Monotonic Extensions of Logic Programming. NMELP '96, Berlin,
536 Heidelberg, Springer-Verlag (1997) 57–70
- 537 5. Artemov, S.N., Protopopescu, T.: Intuitionistic Epistemic Logic. *Rew. Symb. Logic* **9(2)**
538 (2016) 266–298
- 539 6. Dosen, K.: Intuitionistic double negation as a necessity operator. *Publications de l'Institut*
540 *Mathématique, Nouvelle série* **35(49)** (1984) 15–20

⁹ A forum with no formal proceedings but insightful presentations and lively discussions on
epistemic extensions of logic programming systems.

- 541 7. Glivenko, V.: Sur la logique de M. Brouwer. *Bulletin de la Classe des Sciences* **14** (1928)
542 225–228
- 543 8. Dyckhoff, R.: Contraction-free sequent calculi for intuitionistic logic. *Journal of Symbolic*
544 *Logic* **57**(3) (1992) 795807
- 545 9. Hudelmaier, J.: A PROLOG Program for Intuitionistic Logic. SNS-Bericht-. Universität
546 Tübingen (1988)
- 547 10. Hudelmaier, J.: An $O(n \log n)$ -Space Decision Procedure for Intuitionistic Propositional
548 Logic. *Journal of Logic and Computation* **3**(1) (1993) 63–75
- 549 11. Tarau, P.: A Combinatorial Testing Framework for Intuitionistic Propositional Theorem
550 Provers. In Alferes, J.J., Johansson, M., eds.: *Practical Aspects of Declarative Languages*
551 *- 21th International Symposium, PADL 2019, Lisbon, Portugal, January 14-15, 2019, Pro-*
552 *ceedings. Volume 11372 of Lecture Notes in Computer Science., Springer (2019) 115–132*
- 553 12. Ferrari, M., Fiorentini, C., Fiorino, G.: fcube: An efficient prover for intuitionistic propo-
554 sitional logic. In Fermüller, C.G., Voronkov, A., eds.: *Logic for Programming, Artificial*
555 *Intelligence, and Reasoning, Berlin, Heidelberg, Springer Berlin Heidelberg (2010) 294–301*
- 556 13. Fagin, R., Halpern, J.Y.: Belief, Awareness, and Limited Reasoning: Preliminary Report. In:
557 *Proceedings of the 9th International Joint Conference on Artificial Intelligence - Volume 1.*
558 *IJCAI’85, San Francisco, CA, USA, Morgan Kaufmann Publishers Inc. (1985) 491–501*
- 559 14. de Bruijn, N.G.: Exact finite models for minimal propositional calculus over a finite alphabet.
560 Technical Report 75?WSK?02, Technological University Eindhoven (November 1975)
- 561 15. Jongh, D.D., Hendriks, L., de Lavalette, G.R.R.: Computations in fragments of intuitionistic
562 propositional logic. *J. Autom. Reasoning* **7**(4) (1991) 537–561
- 563 16. Statman, R.: Intuitionistic Propositional Logic is Polynomial-Space Complete. *Theor. Com-*
564 *put. Sci.* **9** (1979) 67–72
- 565 17. Egly, U.: A Polynomial Translation of Propositional S4 into Propositional Intuitionistic
566 Logic. (2007)
- 567 18. Goré, R., Thomson, J.: A Correct Polynomial Translation Of S4 Into Intuitionistic Logic.
568 *The Journal of Symbolic Logic* **84**(2) (2019) 439–451
- 569 19. Muggleton, S.: Inductive logic programming. *New Gen. Comput.* **8**(4) (February 1991)
570 295–318
- 571 20. Shapiro, E.Y.: An algorithm that infers theories from facts. In: *Proceedings of the 7th In-*
572 *ternational Joint Conference on Artificial Intelligence - Volume 1. IJCAI’81, San Francisco,*
573 *CA, USA, Morgan Kaufmann Publishers Inc. (1981) 446–451*
- 574 21. Gelfond, M.: New semantics for epistemic specifications. In Delgrande, J.P., Faber, W.,
575 eds.: *Logic Programming and Nonmonotonic Reasoning - 11th International Conference,*
576 *LPNMR 2011, Vancouver, Canada, May 16-19, 2011. Proceedings. Volume 6645 of Lecture*
577 *Notes in Computer Science., Springer (2011) 260–265*
- 578 22. Kracht, M.: On extensions of intermediate logics by strong negation. *Journal of Philosophi-*
579 *cal Logic* **27**(1) (Feb 1998) 49–73
- 580 23. del Cerro, L.F., Herzig, A., Su, E.I.: Epistemic Equilibrium Logic. In Yang, Q., Wooldridge,
581 M.J., eds.: *Proceedings of the Twenty-Fourth International Joint Conference on Artificial*
582 *Intelligence, IJCAI 2015, Buenos Aires, Argentina, July 25-31, 2015, AAAI Press (2015)*
583 *2964–2970*